

# Generic Programming with Dependent Types

Thorsten Altenkirch, Conor McBride and Peter Morris

School of Computer Science and Information Technology  
University of Nottingham

## 1 Introduction

In these lecture notes we give an overview of recent research on the relationship and interaction between two novel ideas in (functional) programming:

**Generic programming** Generic programming [15, 21] allows programmers to explain how a single algorithm can be instantiated for a variety of datatypes, by computation over each datatype’s structure.

**Dependent types** Dependent types [28, 37] are types containing data which enable the programmer to express properties of data concisely, covering the whole spectrum from conventional uses of types to types-as-specifications and programs-as-proofs.

Our central thesis can be summarized by saying that dependent types provide a convenient basis for generic programming by using *universes*. A universe is basically a type  $U : \star$  which contains names for types and a dependent type, or family,  $El : U \rightarrow \star$  which assigns to every name  $a : U$  the type of its elements  $El\ a : \star$ —we call this the *extension* of the name  $a$ . Historically, universes have been already used by Type Theory to capture the predicative hierarchy of types, first introduced by Russell to prevent set-theoretic paradoxes:

$$\star_0 : \star_1 : \star_2 : \dots \star_i : \star_{i+1} : \dots$$

If we want to avoid chains of  $:$  we can represent this hierarchy as:

$$\begin{aligned} U_i &: \star \\ El_i &: U_i \rightarrow \star \\ u_i &: U_{i+1} \end{aligned}$$

Here  $\star$  plays the role of a superuniverse in which all universes can be embedded, while  $U_i$  is the type of (names of) types at level  $i$ . The operation  $El_i$  assigns to any name of a type at level  $i$  the type of its elements. In particular  $u_i$  is the name of the previous universe at level  $i + 1$  and hence  $El_{i+1}\ u_i = U_i$ .

The predicative hierarchy of universes is necessary to have types of types without running into paradoxes (e.g. by having a type of all types). Here we are interested in the application of universes to programming, which leads to consider a wider variety of smaller universes, less general but more usefully structured than the ones above.

## Related work

The idea to use dependent types for generic programming isn't new: starting with the pioneering work by Pfeifer and Rueß [36] who used the LEGO system as a vehicle for generic programming, the authors of [16] actually introduced a universe containing codes for dependent types. The latter is based on the work by Dybjer and Setzer on Induction-Recursion [18, 19] which can also be understood as universe constructions for non-dependent and dependent types. The first two authors of the present notes presented a universe construction for the first order fragment of Haskell datatypes including nested datatypes in [9] which was motivated by the work on generic Haskell [14, 25].

## Structure of the paper

We start our discourse with a quick introduction to dependently typed programming (section 2) using the language Epigram as a vehicle. Epigram is described in more detail elsewhere, see [32] for a definition of the language with many applications, [11] for a short and more recent overview and [31] for an introductory tutorial. Epigram is not just a language but also an interactive program development system, which is, together with further documentation, available from the Epigram homepage [30].

As a warm up we start with a very small, yet useful universe, the universe of finite types (section 3). The names for types in this universe are particularly simple, they are just the natural numbers.

We soon move to bigger universes which include infinite types in section 4, where we introduce a general technique how to represent universes which contain fixpoints of types — this section is based on [35]. We also discuss the tradeoff between the size of a universe and the number of generic operations it supports.

While the universes above are defined syntactically, we also present a semantic approach based on *Container Types* (section 5), see [1–4]. However, here we will not study the categorical details of containers in detail but restrict ourselves to using them to represent datatypes and generic operations in Epigram.

As an example of a library of generic operations we consider the generic zipper [24] (section 6), which is a useful generic tool when implementing functional programs which use the notion of a position in a data structure. As observed by McBride [29], the zipper is closely related to the notion of the derivative of a datatype, which has many structural similarities to derivatives in calculus. This topic has been explored from a more categorical perspective in [5, 7]; the presentation here is again based on [35].

We close with our conclusions and possible directions for further work (section 7).

All the code contained with these notes is available to download from the Epigram website [12].

## 2 Programming with Dependent Types in Epigram

Epigram is an experimental dependently typed functional language and an interactive program development system. It is based on previous experiences with systems based on Type Theory whose emphasis has been on the representations of proofs like LEGO [26]. The design of the interactive program and proof development environment is heavily influenced by the ALF system [27].

Epigram uses a two-dimensional syntax to represent the types of operators in a natural deduction style. This is particularly useful for presenting programs with dependent types—however, we start with some familiar constructions, e.g. we define the Booleans and the Peano-style natural numbers as follows:

```

data (-----! where (-----! ; (-----!
    ! Bool : * )      ! true : Bool )    ! false : Bool )

                                ( n : Nat !
data (-----! where (-----! ; !-----!
    ! Nat : * )      ! zero : Nat )    ! suc n : Nat )

```

We first give the *formation rules*, `Bool` and `Nat` are types (`*` is the type of types) without any assumptions. We then introduce the *constructors*: `true` and `false` are Booleans; `zero` is a natural number and `suc n` is a natural number, if `n` is one. These declarations correspond to the Haskell datatypes:

```

data Bool = True | False
data Nat = Zero | Succ Nat

```

The natural deduction notation may appear quite verbose for simple definitions, but you don't have to manage the layout for yourself—the editor does it for you. On paper, we prefer L<sup>A</sup>T<sub>E</sub>X to ASCII-art, and we take the liberty of typesetting declarations without the bracket delimiters and semicolon separators.

$$\text{data } \frac{}{\text{Nat} : *} \quad \text{where } \frac{}{\text{zero} : \text{Nat}} \quad \frac{n : \text{Nat}}{\text{suc } n : \text{Nat}}$$

The formation rule may contain assumptions as well, declaring the arguments to type constructors, e.g. in the case of lists:

```

( A : * ! ( a : A ; as : List A !
data !-----! where (-----! ; !-----!
    ! List A : * )      ! nil : List A )    ! cons a as : List A )

```

In L<sup>A</sup>T<sub>E</sub>X this becomes:

$$\text{data } \frac{A : *}{\text{List } A : *} \quad \text{where } \frac{}{\text{nil} : \text{List } A} \quad \frac{a : A \quad as : \text{List } A}{\text{cons } a \text{ } as : \text{List } a}$$

Here the type-valued arguments to the constructors are left implicit. However, as we shall see, the use of implicit arguments becomes more subtle with dependent types. Hence, Epigram offers an explicit notation to indicate where implicit arguments should be expected. The definition above can be spelt out in full:

```

-----
( A : * ! ( a : A ; as : List A !
data !-----! where (-----! ; !-----!
! List A ! ! nil _A : ! ! cons _A a as !
! : * ) ! List A ) ! : List A )
-----

```

Here, the underscore overrides the implicit quantifier to allow the user to specify its value explicitly. In L<sup>A</sup>T<sub>E</sub>X we denote this by subscripting the argument itself, as you can see in the typeset version of the above:

$$\text{data } \frac{A : *}{\text{List } A : *} \text{ where } \frac{A : *}{\text{nil}_A : \text{List } A} \frac{A : * \quad a : A \quad as : \text{List } A}{\text{cons}_A a as : \text{List } A}$$

Under normal circumstances, Epigram’s *elaborator* will be able to infer values for these arguments from the way these constructors are used, following a standard unification-based approach to type inference. Previously, we also omitted the declaration of  $A : *$  in the premise: again, this can be inferred by the elaborator as well, with the natural deduction rule acting like Hindley-Milner ‘let’, implicitly generalising local free variables.

### Defining functions

We define functions using `let` and generate patterns interactively. The text of a program records its construction. E.g. we define boolean `not` by `case` analysis on its first argument:

$$\text{let } \frac{b : \text{Bool}}{\text{not } b : \text{Bool}} \quad \text{not } b \Leftarrow \begin{array}{l} \text{case } b \\ \text{not true} \Rightarrow \text{false} \\ \text{not false} \Rightarrow \text{true} \end{array}$$

Let-declarations use the two-dimensional rule syntax to declare a program’s type: in Epigram, all top-level identifiers must be typed explicitly, but this type information is then propagated into their definitions, leaving them relatively free of annotation. Epigram programs are not sequences of prioritised equations as is traditional [33]. Rather, they are treelike: on the left-hand side, the *machine* presents a ‘programming problem’, e.g., to compute `not b`; on the right-hand side, we explain how to attack the problem in one of two ways:

- ⇐ ‘**by**’ refinement into subproblems, using an *eliminator* like `case b`, above; the machine then generates a bunch of subproblems for us to solve;
- ⇒ ‘**return**’ an answer directly, giving an expression of the appropriate return type, which may well be more specific than the type we started with, thanks to the analysis of the problem into cases.

In ASCII source, this tree structure is made explicit using `{...}` markings: here, we drop these in favour of indentation.

Epigram programs thus explain the strategy by which they compute values: case analysis is one such strategy. By ensuring that the available strategies are total (e.g., `case` eliminators cover all constructors), we guarantee that all programs terminate.<sup>1</sup> To implement a recursive program, we can invoke a `rec` eliminator, capturing the strategy of structural recursion on the nominated argument: recursive calls have to be structurally smaller in this argument. As simple examples, consider addition and multiplication of Peano natural numbers:

$$\begin{array}{l} \text{let } \frac{x, y : \text{Nat}}{\text{plus } x \ y : \text{Nat}} \quad \text{plus } x \ y \Leftarrow \text{rec } x \\ \qquad \qquad \qquad \text{plus } x \ y \Leftarrow \text{case } x \\ \qquad \qquad \qquad \text{plus zero } y \Rightarrow y \\ \qquad \qquad \qquad \text{plus (suc } x) \ y \Rightarrow \text{suc (plus } x \ y) \\ \\ \text{let } \frac{m, n : \text{Nat}}{\text{times } m \ n : \text{Nat}} \quad \text{times } m \ n \Leftarrow \text{rec } m \\ \qquad \qquad \qquad \text{times } m \ n \Leftarrow \text{case } m \\ \qquad \qquad \qquad \text{times zero } n \Rightarrow \text{zero} \\ \qquad \qquad \qquad \text{times (suc } m) \ n \Rightarrow \text{plus } n \ (\text{times } m \ n) \end{array}$$

It should be noted that these programs produce terms in Epigram’s underlying type theory which use only standard elimination constants to perform recursion or case analysis. Programming with `rec` is much more flexible than ‘primitive recursion’: it is straightforward to implement programs with deeper structural recursion, like the Fibonacci function, or lexicographically combined structural recursions, like the Ackermann function.

### Data-type families

So far we haven’t used dependent types explicitly. Dependent types come in *families* [20], indexed by data. A standard example is the family of vectors *indexed by length*:

$$\text{data } \frac{n : \text{Nat} \quad X : \star}{\text{Vec } n \ X} \quad \text{where } \frac{}{\text{vnil} : \text{Vec zero } X} \quad \frac{a : A \quad as : \text{Vec } n \ X}{\text{vcons } a \ as : \text{Vec (suc } n) \ X}$$

Here `Vec n X` is the type of vectors, length  $n$ , of items of type  $X$ .

We can now implement a safe version of the head function, whose type makes it clear that the function can only be applied to non-empty lists:

$$\text{let } \frac{ys : \text{Vec (suc } m) \ Y}{\text{vhead } ys : Y} \quad \text{vhead } ys \Leftarrow \text{case } ys \\ \qquad \qquad \qquad \text{vhead (vcons } y \ ys) \Rightarrow y$$

Note that Epigram does not ask for a `vnil` case—`vnil`’s length, `zero`, does not unify with `suc m`, the length of  $ys$ , so the case cannot ever arise.

<sup>1</sup> The condition that the use of universes has to be stratified is present in the language definition but absent from the current implementation. As a consequence, the machine will accept a bogus non-terminating term based on Girard’s paradox.

More generally, we can implement a function which safely accesses any element of a vector. To do this we first define the family of finite types, with the intention that  $\mathbf{Fin} \ n$  represents the finite set  $0, 1, \dots, n - 1$ , i.e. exactly the positions in a vector of length  $n$ .

$$\underline{\text{data}} \frac{n : \mathbf{Nat}}{\mathbf{Fin} \ n : \star} \quad \underline{\text{where}} \frac{}{\mathbf{fz} : \mathbf{Fin} \ (\text{suc } n)} \frac{i : \mathbf{Fin} \ n}{\mathbf{fs} \ i : \mathbf{Fin} \ (\text{suc } n)}$$

Here  $\mathbf{fz}$  represents the 0 which is present in any non-empty finite set, and  $\mathbf{fs} \ i : \mathbf{Fin} \ (\text{suc } n)$  represents  $i + 1$ , given that  $i : \mathbf{Fin} \ n$ . Note that  $\mathbf{Fin} \ \text{zero}$  is meant to be empty: case analysis on a hypothetical element of  $\mathbf{Fin} \ \text{zero}$  leaves the empty set of patterns. The following table enumerates the elements up to  $\mathbf{Fin} \ 4$ :

$\mathbf{Fin} \ 0$	$\mathbf{Fin} \ 1$	$\mathbf{Fin} \ 2$	$\mathbf{Fin} \ 3$	$\mathbf{Fin} \ 4$	$\dots$
	$\mathbf{fz}_0$	$\mathbf{fz}_1$	$\mathbf{fz}_2$	$\mathbf{fz}_3$	$\dots$
		$\mathbf{fs}_1 \ \mathbf{fz}_0$	$\mathbf{fs}_2 \ \mathbf{fz}_1$	$\mathbf{fs}_3 \ \mathbf{fz}_2$	$\dots$
			$\mathbf{fs}_2 \ (\mathbf{fs}_1 \ \mathbf{fz}_0)$	$\mathbf{fs}_3 \ (\mathbf{fs}_2 \ \mathbf{fz}_1)$	$\dots$
				$\mathbf{fs}_3 \ (\mathbf{fs}_2 \ (\mathbf{fs}_1 \ \mathbf{fz}_0))$	$\dots$

As you can see, each non-empty column contains a copy of the previous column, embedded by  $\mathbf{fs}$ , together with a ‘new’  $\mathbf{fz}$  at the start.

We implement the function  $\mathbf{proj}$  which safely accesses the  $i$ th element of a vector by structural recursion over the vector. We analyse the index given as an element of  $\mathbf{Fin} \ n$  and since both constructors of  $\mathbf{Fin} \ n$  produce elements in  $\mathbf{Fin} \ (\text{suc } m)$  the subsequent analysis of the vector needs only a  $\mathbf{vcons}$  case.

$$\underline{\text{let}} \frac{xs : \mathbf{Vec} \ n \ X \quad i : \mathbf{Fin} \ n}{\mathbf{proj} \ xs \ i : X} \quad \begin{aligned} \mathbf{proj} \ xs \ i &\Leftarrow \underline{\text{rec}} \ xs \\ \mathbf{proj} \ xs \ i &\Leftarrow \underline{\text{case}} \ i \\ \mathbf{proj} \ xs \ \mathbf{fz} &\Leftarrow \underline{\text{case}} \ xs \\ \mathbf{proj} \ (\mathbf{vcons} \ x \ xs) \ \mathbf{fz} &\Rightarrow x \\ \mathbf{proj} \ xs \ (\mathbf{fs} \ i) &\Leftarrow \underline{\text{case}} \ xs \\ \mathbf{proj} \ (\mathbf{vcons} \ x \ xs) \ (\mathbf{fs} \ i) &\Rightarrow \mathbf{proj} \ xs \ i \end{aligned}$$

Let’s look more closely at what just happened. Here’s  $\mathbf{proj}$  again, but with the numeric indices shown as subscripts:

$$\underline{\text{let}} \frac{xs : \mathbf{Vec} \ n \ X \quad i : \mathbf{Fin} \ n}{\mathbf{proj}_n \ xs \ i : X} \quad \begin{aligned} \mathbf{proj}_n \ xs \ i &\Leftarrow \underline{\text{rec}} \ xs \\ \mathbf{proj}_n \ xs \ i &\Leftarrow \underline{\text{case}} \ i \\ \mathbf{proj}_{(\text{suc } n)} \ xs \ (\mathbf{fz}_n) &\Leftarrow \underline{\text{case}} \ xs \\ \mathbf{proj}_{(\text{suc } n)} \ (\mathbf{vcons}_n \ x \ xs) \ (\mathbf{fz}_n) &\Rightarrow x \\ \mathbf{proj}_{(\text{suc } n)} \ xs \ (\mathbf{fs}_n \ i) &\Leftarrow \underline{\text{case}} \ xs \\ \mathbf{proj}_{(\text{suc } n)} \ (\mathbf{vcons}_n \ x \ xs) \ (\mathbf{fs}_n \ i) &\Rightarrow \mathbf{proj}_n \ xs \ i \end{aligned}$$

When we analyse  $i$ , we get patterns for  $i$ , but we learn more about  $n$  at the same time. Case analysis specialises the whole programming problem, propagating the consequences of inspecting one value for others related to it by type dependency. The extra requirements imposed on the construction of dependently typed data become extra guarantees when we take it apart.

*Exercise 1.* Implement the function **transpose** which turns an  $m \times n$  matrix represent as an  $m$  vector of  $n$  vectors into an  $n \times m$  matrix represented as a  $n$  vector of  $m$  vectors:

$$\text{let } \frac{xy : \text{Vec } n (\text{Vec } m X)}{\text{transpose } xy : \text{Vec } m (\text{Vec } n X)}$$

## Predefined types in Epigram

Epigram provides very few predefined types: the empty type **Zero**, the unit type **One** and the equality type  $a = b : \star$  for any  $a, b$  not necessarily of the same type. The only constructor for equality is **refl** :  $a = a$  in the special case that both sides of the equation compute to the same value. For example,

$$\text{refl} : \text{plus} (\text{suc} (\text{suc zero})) (\text{suc} (\text{suc zero})) = \text{suc} (\text{suc} (\text{suc} (\text{suc zero})))$$

Epigram has dependent function types  $\forall a : A \Rightarrow B$ , where firstly  $A : \star$  and secondly  $B : \star$  under the assumption  $a : A$ . We retain the conventional  $A \rightarrow B$  notation for function types where the latter assumption is not used. Lambda abstraction is written  $\lambda x : A \Rightarrow b$ . The domain information can be omitted in  $\lambda$  and  $\forall$ , if it can be inferred from the context. Several abstractions of the same kind can be combined using  $;$ , i.e. we write  $\lambda x; y \Rightarrow c$  for  $\lambda x \Rightarrow \lambda y \Rightarrow c$ . The rule notation is just a convenient way to declare functions, e.g. the type of **vhead** can be written explicitly as  $\forall m; Y \Rightarrow \text{Vec} (\text{suc } m) Y \rightarrow Y$ .

## Common datatypes in this paper

There are a few additional standard type constructors which we shall use in this paper: we define a type for disjoint union corresponding to **Either** in Haskell:

$$\text{data } \frac{A, B : \star}{\text{Plus } A B : \star} \quad \text{where } \frac{a : A}{\text{Inl } a : \text{Plus } A B} \quad \frac{b : B}{\text{Inr } b : \text{Plus } A B}$$

Epigram hasn't currently a predefined product type, hence we define it:

$$\text{data } \frac{A, B : \star}{\text{Times } A B} \quad \text{where } \frac{a : A \quad b : B}{\text{Pair } a b : \text{Times } A B}$$

We also introduce the  $\Sigma$ -type, giving us *dependent* tupling:<sup>2</sup>

$$\text{data } \frac{A : \star \quad B : A \rightarrow \star}{\text{Sigma } A B : \star} \quad \text{where } \frac{a : A \quad b : B a}{\text{Tup } a b : \text{Sigma } A B}$$

*Exercise 2.* Define the first and second projection for  $\Sigma$ -types using pattern matching:


$$\text{let } \frac{p : \text{Sigma } A B}{\text{fst } p : A} \quad \text{let } \frac{p : \text{Sigma } A B}{\text{snd } p : B (\text{fst } p)}$$

## W-types

Later in the paper, we shall also need a general-purpose inductive datatype, abstracting once and for all over *well-founded* tree-like data. Tree-like data is built from nodes. Each node carries some sort of data—its *shape*—usually, a tag indicating what sort of node it is, plus some appropriate labelling. In any case, the node shape determines what *positions* there might be for subtrees. This characterisation of well-founded data in terms of shapes and positions is presented via the **W**-type:

$$\text{data } \frac{S : \star \quad P : S \rightarrow \star}{\text{W } S P : \star} \quad \text{where } \frac{s : S \quad f : P s \rightarrow \text{W } S P}{\text{Sup } s f : \text{W } S P}$$

The constructor packs up a choice  $s$  of shape, together with a function  $f$  assigning subtrees to the positions appropriate to that shape. It is traditional to call the constructor **Sup** for ‘supremum’, as a node is the least thing bigger than its subtrees. We often illustrate this pattern—choice of shape, function from positions—as a *triangle* diagram. We write the shape  $s$  in the apex, and we think of the base as the corresponding set  $P s$  of positions. The function  $f$ , part of the node hence inside the triangle, attaches subtrees  $w$  to positions  $p$ .



For example, the natural numbers have *two* node shapes, **suc** with one subtree and **zero** without. Correspondingly, we can use **Bool** for the shapes; the positions

<sup>2</sup> Some authors call this a ‘dependent product’, as it’s the dependent version of **Times**. Other call it a ‘dependent sum’, as it’s the infinitary analogue of **Plus**, and say ‘dependent product’ for functions, as these are the infinitary analogue of tuples. To avoid confusion, we prefer to talk of ‘dependent function types’ and ‘dependent tuple types’.



are given by the following type family which captures ‘being **true**’:

$$\text{data } \frac{b : \text{Bool}}{\text{So } b : \star} \quad \text{where } \frac{}{\text{oh} : \text{So true}}$$

$$\text{let } \frac{p : \text{So false}}{\text{notSo } p : X} \quad \text{notSo } p \leftarrow \text{case } p$$

We may now *define* the natural numbers as a **W**-type:

$$\text{let } \frac{}{\text{wNat} : \star}$$

$$\text{wNat} \Rightarrow \text{W Bool So}$$


$$\text{let } \frac{}{\text{wZero} : \text{wNat}}$$

$$\text{wZero} \Rightarrow \text{Sup false notSo}$$

$$\text{let } \frac{n : \text{wNat}}{\text{wSuc } n : \text{wNat}}$$

$$\text{wSuc } n \Rightarrow \text{Sup true } (\lambda p \Rightarrow n)$$

The pictures show us the components we can plug together to make numbers. ‘Two’ looks like this:



Lots of our favourite inductive datatypes fit this pattern. Another key example is the type of *finitely branching trees*, **W Nat Fin**, where the shape of each node is its arity.

*Exercise 3.* Construct the **W**-type corresponding to Haskell’s

```
data BTree l n = Leaf l | Node (BTree l n) n (BTree l n)
```

Later, we shall exploit the **W**-type analysis of data in terms of shapes and positions to characterise *containers*, more generally.

## Views in Epigram

Once we have the idea of programming by stepwise refinement of problems, it becomes interesting to ask ‘What refinements can we have? Are we restricted to case and rec?’. The *eliminators* which we use to refine programming problems are first-class Epigram values, so it is entirely possible to implement your own. This flexibility is central to the design of Epigram [32], and it gives rise to a novel and useful programming technique inspired by Wadler’s notion of ‘views’ [39].

We can specify a new way to analyse data, just by indexing a datatype family with it. Consider pairs of Boolean values, for example: regardless of whether they

are **true** or **false**, it is surely the case that either they coincide, or the second is the negation of the first. We can express this idea by defining a datatype family—the *view relation*—with one constructor for each case of our desired analysis:

$$\underline{\text{data}} \frac{b, a : \text{Bool}}{\text{EqOrNot } b \ a : \star} \quad \underline{\text{where}} \frac{}{\text{same} : \text{EqOrNot } b \ b \quad \text{diff} : \text{EqOrNot } b \ (\text{not } b)}$$

If we had an element  $p$  of  $\text{EqOrNot } b \ a$ , then case analysis for  $p$  as **same** or **diff** tells us *ipso facto* whether  $a$  is  $b$  or **not**  $b$ . Let us make sure that we can always have such a  $p$  by writing a *covering function* to show that the view relation always holds.

$$\underline{\text{let}} \frac{}{\text{eqOrNot } b \ a : \text{EqOrNot } b \ a} \quad \begin{array}{l} \text{eqOrNot } b \ a \Leftarrow \underline{\text{case } b} \\ \text{eqOrNot } \text{true} \ a \Leftarrow \underline{\text{case } a} \\ \text{eqOrNot } \text{true} \ \text{true} \Rightarrow \text{same} \\ \text{eqOrNot } \text{true} \ \text{false} \Rightarrow \text{diff} \\ \text{eqOrNot } \text{false} \ a \Leftarrow \underline{\text{case } a} \\ \text{eqOrNot } \text{false} \ \text{true} \Rightarrow \text{diff} \\ \text{eqOrNot } \text{false} \ \text{false} \Rightarrow \text{same} \end{array}$$

How do we use this information in practice? Epigram has syntactic support for case analysis derived in this style: if  $p$  is a proof that the view relation holds,  $\underline{\text{view } p}$  is the eliminator which delivers the corresponding analysis of its indices. For example, we may now write

$$\underline{\text{let}} \frac{x, y : \text{Bool}}{\text{xor } x \ y : \text{Bool}} \quad \begin{array}{l} \text{xor } x \ y \Leftarrow \underline{\text{view}} (\text{eqOrNot } x \ y) \\ \text{xor } x \ x \Rightarrow \text{false} \\ \text{xor } x \ (\text{not } x) \Rightarrow \text{true} \end{array}$$

There is no need to be alarmed at the appearance of repeated pattern variables and even *defined* functions on the left-hand side. Operationally, this program computes an element of  $\text{EqOrNot } x \ y$ , then forks control accordingly as it is **same** or **diff**. What you see on the left comes from the specialisation of  $y$  which accompanies that constructor analysis.

Views are important tools for testing data on which types depend. Our  $\text{EqOrNot}$  construction may be more complex than the ordinary Boolean ‘equivalence’ test, but it is also more revealing. The view actually shows the typechecker what the equality test learns.

To see this in action, consider implementing an equality test for **wNat**. At each node, we shall need to compare shapes, and if they coincide, check equality at each position. How do we know that the position sets must be identical

whenever the shapes coincide? Our view makes the connection.

$$\begin{aligned} & \underline{\text{let}} \frac{x, y : \mathbf{wNat}}{\mathbf{wNatEq} \ x \ y : \mathbf{Bool}} \\ & \mathbf{wNatEq} \ x \ y \Leftarrow \underline{\text{rec}} \ x \\ & \mathbf{wNatEq} \ x \ y \Leftarrow \underline{\text{case}} \ x \\ & \mathbf{wNatEq} \ (\text{Sup } b \ f) \ y \Leftarrow \underline{\text{case}} \ y \\ & \mathbf{wNatEq} \ (\text{Sup } b \ f) \ (\text{Sup } a \ g) \Leftarrow \underline{\text{view}} \ (\text{eqOrNot } b \ a) \\ & \mathbf{wNatEq} \ (\text{Sup } b \ f) \ (\text{Sup } b \ g) \Leftarrow \underline{\text{case}} \ b \\ & \mathbf{wNatEq} \ (\text{Sup } \text{true} \ f) \ (\text{Sup } \text{true} \ g) \Rightarrow \mathbf{wNatEq} \ (f \ \text{oh}) \ (g \ \text{oh}) \\ & \mathbf{wNatEq} \ (\text{Sup } \text{false} \ f) \ (\text{Sup } \text{false} \ g) \Rightarrow \text{true} \\ & \mathbf{wNatEq} \ (\text{Sup } b \ f) \ (\text{Sup } (\text{not } b) \ g) \Rightarrow \text{false} \end{aligned}$$

### 3 The Universe of Finite Types

We have already implicitly introduced our first example of a universe: the universe of finite types. The names of finite types are the natural numbers which tell us how many elements the type has and the extension of such a type name is given by the family  $\mathbf{Fin}$  given in the previous section, which assigns to any  $n : \mathbf{Nat}$  a type  $\mathbf{Fin} \ n$  with exactly  $n$  elements. We will now identify basic operations on types within this universe, namely coproducts  $(0, +)$ , products  $(1, \times)$  and leave exponentials  $(\rightarrow)$  as an exercise. This reflects the well known fact that the category of finite types is bicartesian closed.

#### Coproducts

The coproduct of two finite types  $m, n : \mathbf{Nat}$  is simply their arithmetical sum  $\mathbf{plus} \ m \ n : \mathbf{Nat}$ , which we have defined previously. Coproducts come with injections and an eliminator which gives us case analysis. We will use Epigram's views to implement a view on coproducts in the finite universe. As a consequence we can use Epigram's pattern matching to analyse elements of  $\mathbf{Fin} \ (\mathbf{plus} \ m \ n)$  as if they were elements of an ordinary top-level coproduct ( $\mathbf{Plus}$ ).

We are going to parametrize the injections  $\mathbf{finl}$  and  $\mathbf{finr}$  explicitly with the type parameters  $m, n : \mathbf{Nat}$  leading to the following signatures:

$$\underline{\text{let}} \frac{m, n : \mathbf{Nat} \quad i : \mathbf{Fin} \ m}{\mathbf{finl} \ m \ n \ i : \mathbf{Fin} \ (\mathbf{plus} \ m \ n)} \quad \underline{\text{let}} \frac{m, n : \mathbf{Nat} \quad j : \mathbf{Fin} \ n}{\mathbf{finr} \ m \ n \ j : \mathbf{Fin} \ (\mathbf{plus} \ m \ n)}$$

Intuitively,  $\mathbf{finl}$  will map the elements of  $\mathbf{Fin} \ m$  to the first  $m$  elements of  $\mathbf{Fin} \ (\mathbf{plus} \ m \ n)$  and  $\mathbf{finr}$  will map the elements of  $\mathbf{Fin} \ n$  to the subsequent  $n$  elements of  $\mathbf{Fin} \ (\mathbf{plus} \ m \ n)$ . These ideas can be turned into structural recursive programs over  $m$ : in the case of  $\mathbf{finl}$

$$\begin{aligned} \mathbf{finl} \ m \ n \ i & \Leftarrow \underline{\text{rec}} \ m \\ \mathbf{finl} \ m \ n \ i & \Leftarrow \underline{\text{case}} \ i \\ \mathbf{finl} \ (\text{suc } m) \ n \ \text{fz} & \Rightarrow \text{fz} \\ \mathbf{finl} \ (\text{suc } m) \ n \ (\text{fs } i) & \Rightarrow \text{fs} \ (\mathbf{finl} \ m \ n \ i) \end{aligned}$$

we analyse the element  $i : \text{Fin } m$  mapping the constructors **fs**, **fz** in  $\text{Fin } m$  to their counterparts in  $\text{Fin } (\text{plus } m \ n)$ . To implement **finr** we follow a different strategy:

$$\begin{aligned} \text{finr } m \ n \ j &\leftarrow \underline{\text{rec}} \ m \\ \text{finr } m \ n \ j &\leftarrow \underline{\text{case}} \ m \\ \text{finr } \text{zero} \ n \ j &\Rightarrow j \\ \text{finr } (\text{suc } m) \ n \ j &\Rightarrow \text{fs } (\text{finr } m \ n \ j) \end{aligned}$$

We analyse the type name  $m : \text{Nat}$  to apply  $m$  successor operations **fs** to lift  $\text{Fin } n$  into  $\text{Fin } (\text{plus } m \ n)$ . It is worthwhile to note that the above implementations of **finl** and **finr** only work for the given implementation of **plus** which recurs over the first argument. Had we chosen a different one, we would have to either have chosen a different implementation of **finl** and **finr** or would have to employ equational reasoning to justify our implementation. We tend to avoid the latter as much as possible by carefully choosing the way we implement our functions.

How can we compute with elements of  $\text{Fin } (\text{plus } m \ n)$ ? One way to answer this question is to provide an eliminator in form of a case-function:

$$\underline{\text{let}} \frac{s : \text{Fin } (\text{plus } m \ n) \quad l : \text{Fin } m \rightarrow X \quad r : \text{Fin } n \rightarrow X}{\text{fcase}_{m \ n} \ s \ l \ r : X}$$

However, Epigram offers a general mechanism which allows the user to extend the predefined pattern matching mechanism by providing a view, i.e. an alternative covering of a given type which is represented as a family:

$$\begin{aligned} \underline{\text{data}} \quad &\frac{i : \text{Fin } (\text{plus } m \ n)}{\text{FinPlus } m \ n \ i} \quad \underline{\text{where}} \\ &\frac{i : \text{Fin } m}{\text{isfinl } i : \text{FinPlus } m \ n \ (\text{finl } m \ n \ i)} \quad \frac{j : \text{Fin } n}{\text{isfinr } j : \text{FinPlus } m \ n \ (\text{finr } m \ n \ j)} \end{aligned}$$

To use the **FinPlus** view for pattern matching we have to implement a function which witnesses that the covering is exhaustive:

$$\begin{aligned} \underline{\text{let}} \quad &\underline{\text{finPlus}} \ m \ n \ i : \text{FinPlus } m \ n \ i \\ &\text{finPlus } m \ n \ i \leftarrow \underline{\text{rec}} \ m \\ &\text{finPlus } m \ n \ i \leftarrow \underline{\text{case}} \ m \\ &\text{finPlus } \text{zero} \ n \ i \Rightarrow \text{isfinr } i \\ &\text{finPlus } (\text{suc } m) \ n \ i \leftarrow \underline{\text{case}} \ i \\ &\text{finPlus } (\text{suc } m) \ n \ \text{fz} \Rightarrow \text{isfinl } \text{fz} \\ &\text{finPlus } (\text{suc } m) \ n \ (\text{fs } i) \leftarrow \underline{\text{view}} \ \text{finPlus } m \ n \ i \\ &\text{finPlus } (\text{suc } m) \ n \ (\text{fs } (\text{finl } m \ n \ i)) \Rightarrow \text{isfinl } (\text{fs } i) \\ &\text{finPlus } (\text{suc } m) \ n \ (\text{fs } (\text{finr } m \ n \ j)) \Rightarrow \text{isfinr } j \end{aligned}$$

We can now use **view** to do pattern matching over  $\text{Fin } (\text{plus } m \ n)$ , e.g. to implement **fcase**:

$$\begin{aligned} \text{fcase}_{m \ n} \ s \ l \ r &\leftarrow \underline{\text{view}} \ \text{finPlus } m \ n \ s \\ \text{fcase}_{m \ n} \ (\text{finl } m \ n \ i) \ l \ r &\Rightarrow l \ i \\ \text{fcase}_{m \ n} \ (\text{finr } m \ n \ j) \ l \ r &\Rightarrow r \ j \end{aligned}$$

## Products

Given type names  $m, n : \mathbf{Nat}$  their cartesian product is denoted by the arithmetic product  $\mathbf{times} \ m \ n$ . Elements of  $\mathbf{Fin} \ (\mathbf{times} \ m \ n)$  can be constructed using pairing:

$$\underline{\text{let}} \frac{i : \mathbf{Fin} \ m \quad j : \mathbf{Fin} \ n}{\mathbf{fpair} \ m \ n \ i \ j : \mathbf{Fin} \ (\mathbf{times} \ m \ n)}$$

The intuitive idea is to arrange the elements of  $\mathbf{Fin} \ (\mathbf{times} \ m \ n)$  as a rectangle and assign to  $\mathbf{pair} \ i \ j$  the  $j$ th column in the  $i$ th row. This is realised by the following primitive recursive function which uses the previously defined constructors for coproducts, since our products are merely iterated coproducts:

$$\begin{aligned} \mathbf{fpair} \ m \ n \ i \ j &\Leftarrow \underline{\text{rec}} \ i \\ \mathbf{fpair} \ m \ n \ i \ j &\Leftarrow \underline{\text{case}} \ i \\ \mathbf{fpair} \ (\text{suc } m) \ n \ \text{fz } j &\Rightarrow \mathbf{finl} \ n \ (\mathbf{times} \ m \ n) \ j \\ \mathbf{fpair} \ (\text{suc } m) \ n \ (\text{fs } i) \ j &\Rightarrow \mathbf{finr} \ n \ (\mathbf{times} \ m \ n) \ (\mathbf{fpair} \ m \ n \ i \ j) \end{aligned}$$

Indeed the  $\mathbf{pair} \ m \ n \ i \ j$  just computes  $j + i * n$ , however our implementation verifies that the result is less than  $m * n$  simply by type checking.

As in the case for coproducts we extend pattern matching to cover our products by providing the appropriate view:

$$\begin{aligned} \underline{\text{data}} \quad & \frac{i : \mathbf{Fin} \ (\mathbf{times} \ m \ n)}{\mathbf{FinTimes} \ m \ n \ i : \star} \\ \underline{\text{where}} \quad & \frac{i : \mathbf{Fin} \ m \quad j : \mathbf{Fin} \ n}{\mathbf{isfpair} \ i \ j : \mathbf{FinTimes} \ m \ n \ (\mathbf{fpair} \ m \ n \ i \ j)} \end{aligned}$$

As before we show that this view is exhaustive:

$$\begin{aligned} \underline{\text{let}} \quad & \frac{}{\mathbf{finTimes} \ m \ n \ i : \mathbf{FinTimes} \ m \ n \ i} \\ \mathbf{finTimes} \ m \ n \ i &\Leftarrow \underline{\text{rec}} \ m \\ \mathbf{finTimes} \ m \ n \ i &\Leftarrow \underline{\text{case}} \ m \\ \mathbf{finTimes} \ \text{zero} \ n \ i &\Leftarrow \underline{\text{case}} \ i \\ \mathbf{finTimes} \ (\text{suc } m) \ n \ i &\Leftarrow \underline{\text{view}} \ \mathbf{finPlus} \ n \ (\mathbf{times} \ m \ n) \ i \\ \mathbf{finTimes} \ (\text{suc } m) \ n \ (\mathbf{finl} \ n \ (\mathbf{times} \ m \ n) \ i) &\Rightarrow \mathbf{isfpair} \ \text{fz } i \\ \mathbf{finTimes} \ (\text{suc } m) \ n \ (\mathbf{finr} \ n \ (\mathbf{times} \ m \ n) \ j) &\Leftarrow \underline{\text{view}} \ \mathbf{finTimes} \ m \ n \ j \\ \mathbf{finTimes} \ (\text{suc } m) \ n \ (\mathbf{finr} \ n \ (\mathbf{times} \ m \ n) \ (\mathbf{fpair} \ m \ n \ i \ j)) &\Rightarrow \mathbf{isfpair} \ (\text{fs } i) \ j \end{aligned}$$

Note that we are using the previously defined  $\mathbf{FinPlus}$  view to analyse the iterated coproducts. We can use both derived pattern matching principles to

show that products distribute over coproducts

$$\begin{aligned}
& \underline{\text{let}} \frac{x : \text{Fin}(\text{times } m (\text{plus } n o))}{\text{dist } m n o x : \text{Fin}(\text{plus}(\text{times } m n)(\text{times } m o))} \\
& \text{dist } m n o x \Leftarrow \underline{\text{view}} \text{finTimes } m (\text{plus } n o) x \\
& \text{dist } m n o (\text{fpair } m (\text{plus } n o) i j) \Leftarrow \underline{\text{view}} \text{finPlus } n o j \\
& \quad \text{dist } m n o (\text{fpair } m (\text{plus } n o) i (\text{finl } n o j)) \\
& \quad \Rightarrow \text{finl}(\text{times } m n)(\text{times } m o) (\text{fpair } m n i j) \\
& \quad \text{dist } m n o (\text{fpair } m (\text{plus } n o) i (\text{finr } n o j)) \\
& \quad \Rightarrow \text{finr}(\text{times } m n)(\text{times } m o) (\text{fpair } m o i j)
\end{aligned}$$

The categorically inclined may notice that this is not an automatic consequence of having products and coproducts, but usually established as a consequence of having exponentials. We leave it as an exercise to define exponentials.

*Exercise 4.* Define exponentials (i.e. function types) by implementing a function to represent the name of a function type:

$$\underline{\text{let}} \frac{m, n : \text{Nat}}{\text{exp } m n : \text{Nat}}$$

and a constructor corresponding to lambda abstraction:

$$\underline{\text{let}} \frac{f : \text{Fin } m \rightarrow \text{Fin } n}{\text{flam } m n f : \text{Fin}(\text{exp } m n)}$$

Unlike in the previous cases we cannot implement a pattern matching principle due to the lack of extensionality in Epigram's type system.<sup>3</sup>

However, we can define an application operator:

$$\underline{\text{let}} \frac{f : \text{Fin}(\text{exp } m n) \quad i : \text{Fin } m}{\text{fapp } m n f i : \text{Fin } n}$$

## 4 Universes for Generic Programming

The previously introduced universe of finite types is extensional, any two functions which are extensionally equal are given the same code. E.g. using the example from [13] we can see that the functions  $\lambda f : \text{Bool} \rightarrow \text{Bool} \Rightarrow f$  and  $\lambda f : \text{Bool} \rightarrow \text{Bool}; x : \text{Bool} \Rightarrow f(f(f x))$  are extensionally equal by encoding them

<sup>3</sup> We cannot show that two functions are equal if they are pointwise equal. As a consequence we cannot show for example that there are exactly 4 functions of type  $\text{Bool} \rightarrow \text{Bool}$  which would be necessary if we want to establish a case analysis principle for finite types. Our ongoing work on *Observational Type Theory* [10] will address this issue.

using the combinators defined in the previous section and observing that they compute the same element in [Fin 256](#).<sup>4</sup>

While extensionality is a desirable feature, it is not always as easy to achieve as in the case of finite types. Hence, when moving to larger universes which allow us to represent infinite datatypes we shall use a different approach. Instead of identifying our type constructors within a given type of names, we inductively define the type of type names and the family of inhabitants.

## Finite types, revisited

To illustrate this let us revisit the universe of finite types, we can inductively define the type names generated from 0, +, 1, ×:

$$\begin{array}{c} \text{data } \frac{}{\mathbf{Ufin} : \star} \quad \text{where } \frac{}{\mathbf{'0'} : \mathbf{Ufin}} \quad \frac{a, b : \mathbf{Ufin}}{\mathbf{'plus'} a b : \mathbf{Ufin}} \\ \frac{}{\mathbf{'1'} : \mathbf{Ufin}} \quad \frac{a, b : \mathbf{Ufin}}{\mathbf{'times'} a b : \mathbf{Ufin}} \end{array}$$

We could also have included function types, however, they will require special attention later when we introduce inductive types.

We define the family of elements [Elfin](#) inductively:

$$\begin{array}{c} \text{data } \frac{a : \mathbf{Ufin}}{\mathbf{Elfin} a : \star} \\ \text{where } \frac{b, a : \mathbf{Ufin} \quad x : \mathbf{Elfin} a}{\mathbf{inl} x : \mathbf{Elfin} (\mathbf{'plus'} a b)} \quad \frac{a, b : \mathbf{Ufin} \quad y : \mathbf{Elfin} b}{\mathbf{inr} y : \mathbf{Elfin} (\mathbf{'plus'} a b)} \\ \frac{}{\mathbf{void} : \mathbf{Elfin} \mathbf{'1'}} \quad \frac{x : \mathbf{Elfin} a \quad y : \mathbf{Elfin} b}{\mathbf{pair} x y : \mathbf{Elfin} (\mathbf{'times'} a b)} \end{array}$$

Indeed, we have seen inductively defined families already when we introduced [Vec](#) and [Fin](#). We can reimplement the [dist](#) function for this universes without having to resort to views, the built in pattern matching will do the job:

$$\begin{array}{c} \text{let } \frac{x : \mathbf{Elfin} (\mathbf{'times'} a (\mathbf{'plus'} b c))}{\mathbf{dist} x : \mathbf{Elfin} (\mathbf{'plus'} (\mathbf{'times'} a b) (\mathbf{'times'} a c))} \\ \mathbf{dist} x \Leftarrow \text{case } x \\ \mathbf{dist} (\mathbf{pair} x y) \Leftarrow \text{case } y \\ \mathbf{dist} (\mathbf{pair} x (\mathbf{inl} y)) \Rightarrow \mathbf{inl} (\mathbf{pair} x y) \\ \mathbf{dist} (\mathbf{pair} x (\mathbf{inr} z)) \Rightarrow \mathbf{inr} (\mathbf{pair} x z) \end{array}$$

<sup>4</sup> We don't recommend trying this with the current implementation of Epigram.

## 4.1 Enumerating finite types

So far we haven't defined any proper generic operations, i.e. an operation which works on all types of a universe by inspecting the name. A generic operation which is *typical* for finite types is the possibility to enumerate all elements of a given type. We shall use binary trees instead of lists to represent the results of an enumeration so that the path in the tree correspond to the choices we have to make to identify the element. Since our types may be empty we require a special constructor to represent an empty tree:

$$\text{data } \frac{A : \star}{\text{ET } A : \star}$$

$$\text{where } \frac{a : A}{\text{V } a : \text{ET } A} \quad \frac{l, r : \text{ET } A}{\text{C } l r : \text{ET } A} \quad \frac{}{\text{E} : \text{ET } A}$$

Our generic enumeration function has the following type:

$$\text{let } \frac{a : \text{Ufin}}{\text{enum } a : \text{ET } (\text{Efin } a)}$$

To implement **enum** it is helpful to observe that **ET** is a monad, with

$$\text{let } \frac{a : A}{\text{returnET } a : \text{ET } A} \quad \text{returnET } a \Rightarrow \text{V } a$$

$$\text{let } \frac{t : \text{ET } A \quad f : A \rightarrow \text{ET } B}{\text{bindET } t f : \text{ET } B}$$

$$\begin{aligned} \text{bindET } t f &\Leftarrow \text{rec } t \\ \text{bindET } t f &\Leftarrow \text{case } t \\ \text{bindET } (\text{V } a) f &\Rightarrow f a \\ \text{bindET } (\text{C } l r) f &\Rightarrow \text{C } (\text{bindET } l f) (\text{bindET } r f) \\ \text{bindET } \text{E} f &\Rightarrow \text{E} \end{aligned}$$

Consequently, **ET** is also functorial:

$$\text{let } \frac{f : A \rightarrow B \quad t : \text{ET } A}{\text{mapET } f t : \text{ET } B}$$

$$\text{mapET } f t \Rightarrow \text{bindET } t (\lambda x \Rightarrow \text{returnET } (f x))$$

We are now ready to implement **enum** by structural recursion over the type name:

$$\begin{aligned} \text{enum } a &\Leftarrow \text{rec } a \\ \text{enum } a &\Leftarrow \text{case } a \\ \text{enum } '0' &\Rightarrow \text{E} \\ \text{enum } ('plus' a b) &\Rightarrow \text{C } (\text{mapET } \text{inl } (\text{enum } a)) (\text{mapET } \text{inr } (\text{enum } b)) \\ \text{enum } '1' &\Rightarrow \text{V void} \\ \text{enum } ('times' a b) &\Rightarrow \text{bindET } (\text{enum } a) (\lambda x \Rightarrow \text{mapET } (\lambda y \Rightarrow \text{pair } x y) (\text{enum } b)) \end{aligned}$$



*Exercise 5.* Add function types to `Ufin` and extend `Elfin`. Can you extend `enum` to cover function types?

### Context-free types

By context-free types<sup>5</sup> we mean types which can be constructed by combining the polynomial operators from the previous section  $(0, +, \times, 1)$  with an operator  $\mu$  to construct inductive types, or in categorical terms initial algebras. We have already seen some examples of context-free types, for instance `Nat` can be expressed as `Nat` =  $\mu X.1 + X$  and `List` which can be encoded: `List`  $A = \mu X.1 + A \times X$ . Other examples we will use include binary trees with data at the nodes which can be given by the expression `Tree`  $A = \mu X.1 + (X \times A \times X)$ . Finally rose trees which are given by the code `RT`  $A = \mu X.\text{List}(A \times X) = \mu X.\mu Y.1 + (A \times X) \times Y$ . We use the term context-free types because the types have the same structure as context-free grammars, identifying parameters with terminal symbols, recursive variables with non-terminal symbols, choice with  $+$  and sequence with  $\times$ .

The first technical issue we need to address is how to represent variables. We use a deBruijn style representation of variables, this seems to be essential since we give are going to represent types an inductive family, using names would cause a considerable overhead and also would mean that we have to deal with issues like alpha conversion. Moreover, we are free to implement a function which translates a name carrying type into our internal deBruijn representation. This choice is a variation on the approach taken by McBride[29] when he first gave an inductive characterisation of these types. The names of context free types becomes a family indexed by the number of free variables:

$$\text{data } \frac{n : \text{Nat}}{\text{Ucf } n : \star} \text{ where}$$

as constructors we retain the polynomial operators which leave the number of free variables unchanged:

$$\frac{}{\text{'0'} : \text{Ucf } n} \quad \frac{a, b : \text{Ucf } n}{\text{'plus'} a b : \text{Ucf } n}$$

$$\frac{}{\text{'1'} : \text{Ucf } n} \quad \frac{a, b : \text{Ucf } n}{\text{'times'} a b : \text{Ucf } n}$$

To represent variables we introduce two constructors: `vl` which represents the last variable in a non-empty context, and `wk a` which means that the type name  $a$  is weakened, i.e. the last variable is not used and `vl` now refers to the variable before the last:

$$\frac{}{\text{vl} : \text{Ucf } (\text{suc } n)} \quad \frac{a : \text{Ucf } n}{\text{wk } a : \text{Ucf } (\text{suc } n)}$$

<sup>5</sup> We previously used the term 'regular tree types' [35] in a vain attempt to avoid confusion with regular *expressions*.

An alternative is to use the previously defined family of finite types directly, i.e. only to introduce one constructor:

$$\frac{x : \mathbf{Fin} \ n}{\mathbf{var} \ x : \mathbf{Ucf} \ n}$$

but it is slightly more convenient to use **wk** and **vl** because otherwise we have to define operations on **Fin** and **Ucf** instead of just for **Ucf**.

Dual to weakening is an operator representing *local definitions*, which allows us to replace the last variable by a given type name:

$$\frac{f : \mathbf{Ucf} \ (\mathbf{suc} \ n) \quad a : \mathbf{Ucf} \ n}{\mathbf{def} \ f \ a : \mathbf{Ucf} \ n}$$

Alternatively, we could have defined substitution by recursion over the structure of type names. In the presence of a binding operator, here  $\mu$ , this is not completely trivial. Later, we will see that another advantage of local definitions is that it allows us to define operations by structural recursion whose termination would have to be justified otherwise.

Finally, we introduce the constructor for inductive types, which binds the last variable and hence decreases the number of free variables by one:

$$\frac{f : \mathbf{Ucf} \ (\mathbf{suc} \ n)}{\mathbf{'mu'} \ f : \mathbf{Ucf} \ n}$$

Our examples (natural numbers, lists, trees and rose trees) can be translated into type names in **Ucf**:

$$\mathbf{let} \ \frac{}{\mathbf{nat} : \mathbf{Ucf} \ n} \quad \mathbf{nat} \Rightarrow \mathbf{'mu'} \ (\mathbf{'plus'} \ \mathbf{'1'} \ \mathbf{vl})$$

$$\mathbf{let} \ \frac{}{\mathbf{list} : \mathbf{Ucf} \ (\mathbf{suc} \ n)} \quad \mathbf{list} \Rightarrow \mathbf{'mu'} \ (\mathbf{'plus'} \ \mathbf{'1'} \ (\mathbf{'times'} \ (\mathbf{wk} \ \mathbf{vl}) \ \mathbf{vl}))$$

$$\mathbf{let} \ \frac{}{\mathbf{tree} : \mathbf{Ucf} \ (\mathbf{suc} \ n)} \\ \mathbf{tree} \Rightarrow \mathbf{'mu'} \ (\mathbf{'plus'} \ \mathbf{'1'} \ (\mathbf{'times'} \ \mathbf{vl} \ (\mathbf{'times'} \ (\mathbf{wk} \ \mathbf{vl}) \ \mathbf{vl})))$$

$$\mathbf{let} \ \frac{}{\mathbf{rt} : \mathbf{Ucf} \ (\mathbf{suc} \ n)} \quad \mathbf{rt} \Rightarrow \mathbf{'mu'} \ (\mathbf{def} \ \mathbf{list} \ (\mathbf{'times'} \ (\mathbf{wk} \ \mathbf{vl}) \ \mathbf{vl}))$$

While **nat** and **rt** are closed types and hence inhabit **Ucf**  $n$  for any  $n : \mathbf{Nat}$ , **list** is parametrized by the last type variable and hence inhabits **Ucf** (**suc**  $n$ ). We exploit this in the definition of rose trees where we construct rose trees as the initial algebra of **list**. Alternatively we can instantiate **list** to any type name using **let**, e.g. **let list nat : Ucf**  $n$  represents the type of lists of natural numbers.

## 4.2 Elements of context-free types

How are we going to define the family of elements for **Ucf**? We have to take care of the free type variables. A first attempt would be to say that we have to interpret any type variable by a type, leading to the following signature<sup>6</sup> :

$$\frac{a : \mathbf{Ucf} \ n \ Xs : \mathbf{Vec} \ n \ \star}{\mathbf{Elcf} \ a \ Xs : \star}$$

This approach works fine for the polynomial operators, which are interpreted as before, and the variables which correspond to projections; however, we run in difficulties for  $\mu$ . Let's see why: A reasonable attempt is to say:

$$\frac{x : \mathbf{Elcf} \ f \ (\mathbf{vcons} \ (\mathbf{Elcf} \ ('mu' \ f) \ Xs) \ Xs)}{\mathbf{in} \ x : \mathbf{Elcf} \ ('mu' \ f) \ Xs}$$

However, this definition is not accepted by Epigram's schema checker, since it is not able to verify that the nested occurrence of **Elcf** is only used in a strictly positive fashion. This check is necessary to keep Epigram's type system sound by avoiding potentially non-terminating programs.

However, if we restrict ourselves to interpreting only closed types, we can overcome this problem. We define the **Elcf** wrt to a *closing substitution* or telescope, which interprets any free type variable by a type name with fewer free variables. Hence we define the family of telescopes:

$$\mathbf{data} \ \frac{n : \mathbf{Nat}}{\mathbf{Tel} \ n : \star} \ \mathbf{where} \ \frac{}{\mathbf{tnil} : \mathbf{Tel} \ \mathbf{zero}} \ \frac{a : \mathbf{Ucf} \ n \ as : \mathbf{Tel} \ n}{\mathbf{tcons} \ a \ as : \mathbf{Tel} \ (\mathbf{suc} \ n)}$$

We can now define an interpretation for an open type together with a fitting telescope:

$$\mathbf{data} \ \frac{a : \mathbf{Ucf} \ n \ as : \mathbf{Tel} \ n}{\mathbf{Elcf} \ a \ as : \star}$$

the constructors for the polynomial operators stay the same, only indexed with a telescope which is passed through:

$$\frac{b, a : \mathbf{Ucf} \ x : \mathbf{Elcf} \ a \ as}{\mathbf{inl} \ x : \mathbf{Elcf} \ ('plus' \ a \ b) \ as} \quad \frac{a, b : \mathbf{Ucf} \ y : \mathbf{Elcf} \ b \ as}{\mathbf{inr} \ y : \mathbf{Elcf} \ ('plus' \ a \ b) \ as}$$

$$\frac{}{\mathbf{void} : \mathbf{Elcf} \ '1' \ as} \quad \frac{x : \mathbf{Elcf} \ a \ y : \mathbf{Elcf} \ b \ as}{\mathbf{pair} \ x \ y : \mathbf{Elcf} \ ('times' \ a \ b) \ as}$$

The interpretation of the last variable is simply the interpretation of the first type name in the telescope:

$$\frac{x : \mathbf{Elcf} \ a \ as}{\mathbf{top} \ x : \mathbf{Elcf} \ \mathbf{vl} \ (\mathbf{tcons} \ a \ as)}$$

<sup>6</sup> We are exploiting here  $\star : \star$ , however, this use can be stratified, i.e. if  $Xs : \star_i$  then  $\mathbf{Elcf} \ a \ Xs : \star_{i+1}$ .

Meanwhile, the interpretation of a weakened type is given by popping off the first item of the telescope:

$$\frac{x : \text{Elcf } a \text{ } as}{\text{pop } x : \text{Elcf } (\text{wk } a) (\text{tcons } b \text{ } as)}$$

A local definition is explained by pushing the right hand side of the definition onto the telescope stack:

$$\frac{x : \text{Elcf } f (\text{tcons } a \text{ } as)}{\text{push } x : \text{Elcf } (\text{def } f \text{ } a) \text{ } as}$$

We can finally reap the fruits of our syntactic approach by providing an interpretation of **mu** which doesn't require a nested use of **Elcf**:

$$\frac{x : \text{Elcf } f (\text{tcons } as ('mu' f))}{\text{in } x : \text{Elcf } ('mu' f) \text{ } as}$$

We can now derive constructors for our encoded types and provide a derived case analysis using views. We show this in the case of **nat** and leave the other examples as an exercise.

We derive the constructors representing 0 and successor:

$$\text{let } \frac{}{\text{'zero' } : \text{Elcf } \text{nat } as} \quad \text{'zero' } \Rightarrow \text{in } (\text{inl void})$$

$$\text{let } \frac{n : \text{Elcf } \text{nat } as}{\text{'suc' } n : \text{Elcf } \text{nat } as} \quad \text{'suc' } n \Rightarrow \text{in } (\text{inr } (\text{top } n))$$

Our view is that all elements of **Elcf nat** are constructed by one of the constructors, this is expressed by the family **NatView**:

$$\text{data } \frac{n : \text{Elcf } \text{nat } as}{\text{NatView } n : \star}$$

$$\text{where } \frac{}{\text{isZ } : \text{NatView } \text{'zero' }} \quad \frac{n : \text{Elcf } \text{nat } as}{\text{isS } n : \text{NatView } (\text{'suc' } n)}$$

We show that **NatView** is exhaustive:

$$\begin{aligned} \text{let } & \frac{n : \text{Elcf } \text{nat } as}{\text{natView } n : \text{NatView } n} \\ \text{natView } n & \Leftarrow \text{case } n \\ \text{natView } (\text{in } x) & \Leftarrow \text{case } x \\ \text{natView } (\text{in } (\text{inl } x)) & \Leftarrow \text{case } x \\ \text{natView } (\text{in } (\text{inl void})) & \Rightarrow \text{isZ} \\ \text{natView } (\text{in } (\text{inr } y)) & \Leftarrow \text{case } y \\ \text{natView } (\text{in } (\text{inr } (\text{top } n'))) & \Rightarrow \text{isS } n' \end{aligned}$$

We can now use the derived pattern matching principle to implement functions over the encoded natural numbers:

$$\begin{aligned} \underline{\text{let}} \quad & \frac{m, n : \text{Elcf nat } as}{\text{'add'} m n : \text{Elcf nat } as} \\ & \text{'add'} m n \Leftarrow \underline{\text{rec}} m \\ & \text{'add'} m n \Leftarrow \underline{\text{view}} \text{ natView } m \\ & \text{'add'} (\text{in } (\text{inl void})) n \Rightarrow n \\ & \text{'add'} (\text{in } (\text{inr } (\text{top } m'))) n \Rightarrow \text{in } (\text{inr } (\text{top } (\text{'add'} m' n))) \end{aligned}$$

Unfortunately, epigram always normalizes terms which appear in patterns, expanding **'zero'** and **'suc'**, which makes the pattern not very readable.

Note that we don't need to derive a new recursion principle since structural recursion over the encoded natural numbers is the same as structural recursion over the natural numbers.

A natural question is whether we should have to distinguish between *encoded natural numbers* and *natural numbers* at all. The answer is clearly **no**, since they are isomorphic anyway. To exploit this fact and avoid unnecessary duplication of definitions we need to build in a reflection mechanism into Epigram which allows us to access the names of the top level universe as data.

*Exercise 6.* Define a pattern matching principle for lists, that is first define 'constructors'

$$\begin{aligned} \underline{\text{let}} \quad & \frac{}{\text{'nil'} : \text{Elcf list } X} \\ \underline{\text{let}} \quad & \frac{x : \text{Elcf } a \text{ } as \quad xs : \text{Elcf list } (\text{tcons } a \text{ } as)}{\text{'cons'} x xs : \text{Elcf list } (\text{tcons } a \text{ } as)} \end{aligned}$$

and then create an appropriate view, following the nat example. Consider how to do this for rose trees.

The *typical* generic operation on context-free types is generic equality. We can implement generic equality by structural recursion over the elements in `Elcf`:

```

let  $\frac{x, y : \text{Elcf } a \text{ as}}{\text{geq } x \ y : \text{Bool}}$ 
  geq  $x \ y \leftarrow \text{rec } x$ 
  geq  $x \ y \leftarrow \text{case } x$ 
  geq  $(\text{inl } xa) \ y \leftarrow \text{case } y$ 
  geq  $(\text{inl } xa) (\text{inl } ya) \Rightarrow \text{geq } xa \ ya$ 
  geq  $(\text{inl } xa) (\text{inr } yb) \Rightarrow \text{false}$ 
  geq  $(\text{inr } xb) \ y \leftarrow \text{case } y$ 
  geq  $(\text{inr } xb) (\text{inl } ya) \Rightarrow \text{false}$ 
  geq  $(\text{inr } a) (\text{inr } yb) \Rightarrow \text{geq } xb \ yb$ 
  geq  $\text{void } y \leftarrow \text{case } y$ 
  geq  $\text{void void} \Rightarrow \text{true}$ 
  geq  $(\text{pair } xa \ xb) \ y \leftarrow \text{case } y$ 
  geq  $(\text{pair } xa \ xb) (\text{pair } ya \ yb) \Rightarrow \text{and } (\text{geq } xa \ ya) (\text{geq } xb \ yb)$ 
  geq  $(\text{top } x) \ y \leftarrow \text{case } y$ 
  geq  $(\text{top } x) (\text{top } y) \Rightarrow \text{geq } x \ y$ 
  geq  $(\text{pop } x) \ y \leftarrow \text{case } y$ 
  geq  $(\text{pop } x) (\text{pop } y) \Rightarrow \text{geq } x \ y$ 
  geq  $(\text{push } x) \ y \leftarrow \text{case } y$ 
  geq  $(\text{push } x) (\text{push } y) \Rightarrow \text{geq } x \ y$ 
  geq  $(\text{in } x) \ y \leftarrow \text{case } y$ 
  geq  $(\text{in } x) (\text{in } y) \Rightarrow \text{geq } x \ y$ 

```

The algorithm is completely data-driven — indeed we never inspect the type. However, using the type information, the choice of the first argument limits the possible cases for the 2nd. This is what dependently typed pattern matching buys us, that it records the consequences of choices we have already made.

Note that we don't have to assume that the equality of type parameters is decidable. This is due to the fact that we only derive generic operations for closed types here.

*Exercise 7.* Instead of just returning a boolean we can actually show that we can decide equality of elements of `Elcf`. We say that a type is **decided**, if it can be established whether it is empty or inhabited. This is reflected by the following definition:

$$\text{let } \frac{A : \star}{\text{Not } A : \star} \quad \text{Not } A \Rightarrow A \rightarrow \text{Zero}$$

$$\text{data } \frac{A : \star}{\text{Dec } A : \star} \quad \text{where } \frac{a : A}{\text{yes } a : \text{Dec } A} \quad \frac{f : \text{Not } A}{\text{no } f : \text{Dec } A}$$

Here `Zero` is Epigram's built in empty type, establishing a function of type `Not A`, i.e.  $A \rightarrow \text{Zero}$  establishes that  $A$  is uninhabited.

To show that equality for context-free types is decidable we have to implement:

$$\underline{\text{let}} \frac{x, y : \text{Elcf } a \text{ as}}{\text{geqdec } x \ y : \text{Dec } (x = y)}$$

**geqdec** is a non-trivial refinement of **geq** using Epigram's type system to show that the implementation of the program delivers what its name promises.

### 4.3 Strictly positive types

Context-free types capture most of the types which are useful in daily functional programming. However, in some situations we want to use trees which are infinitely branching. E.g. we may want to define a system of ordinal notations, which extends natural numbers by the possibility to form the limit, i.e. the least upper bound, of an infinite sequence of ordinals.

$$\underline{\text{data}} \frac{}{\text{Ord} : \star} \quad \underline{\text{where}} \frac{}{\text{oz} : \text{Ord}} \quad \frac{a : \text{Ord}}{\text{os } a : \text{Ord}} \quad \frac{f : \text{Nat} \rightarrow \text{Ord}}{\text{olim } f : \text{Ord}}$$

We can embed the natural numbers into the ordinals:

$$\underline{\text{let}} \frac{n : \text{Nat}}{\text{o2n } n : \star} \quad \begin{array}{l} \text{n2o } n \Leftarrow \underline{\text{rec}} \ n \\ \text{n2o } n \Leftarrow \underline{\text{case}} \ n \\ \text{n2o } \text{zero} \Rightarrow \text{oz} \\ \text{n2o } (\text{suc } n) \Rightarrow \text{os } (\text{n2o } n) \end{array}$$

and using this embedding we define the first infinite ordinal ( $\omega$ ) as the limit of the sequence of all natural numbers:

$$\underline{\text{let}} \frac{}{\text{omega} : \text{Ord}} \quad \text{omega} \Rightarrow \text{olim } \text{n2o}$$

We can also do arithmetic on ordinals, using structural recursion we define addition<sup>7</sup> of ordinals:

$$\underline{\text{let}} \frac{a, b : \text{Ord}}{\text{oplus } a \ b : \text{Ord}} \quad \begin{array}{l} \text{oplus } a \ b \Leftarrow \underline{\text{rec}} \ b \\ \text{oplus } a \ b \Leftarrow \underline{\text{case}} \ b \\ \text{oplus } a \ \text{oz} \Rightarrow a \\ \text{oplus } a \ (\text{os } b) \Rightarrow \text{os } (\text{oplus } a \ b) \\ \text{oplus } a \ (\text{olim } f) \Rightarrow \text{olim } (\lambda n \Rightarrow \text{oplus } a \ (f \ n)) \end{array}$$

Categorically, **Ord** is an initial algebra  $\mu X.1 + X + \text{Nat} \rightarrow X$ , it is an instance of a *strictly positive type*. Strictly positive types may use function types, like

<sup>7</sup> To reflect the standard definition of ordinal addition we **have** to recur on the 2nd argument. Ordinal addition is not commutative,  $\omega + 1$  denotes the successor of  $\omega$ , while  $1 + \omega$  is order-isomorphic to  $\omega$ .

in  $\mathbf{Nat} \rightarrow X$  but we do not allow type variables to appear on the left-hand side of the arrow. I.e.  $\mu X.X \rightarrow \mathbf{Bool}$  is not strictly positive because  $X$  appears negatively, but neither is  $\mu X.(X \rightarrow \mathbf{Bool}) \rightarrow \mathbf{Bool}$  because  $X$  appears positively but not strictly positive.

We introduce the universe of strictly positive types by amending the universe of context-free types. That is we define

$$\underline{\text{data}} \frac{n : \mathbf{Nat}}{\mathbf{Usp} \ n : \star}$$

$$\underline{\text{data}} \frac{a : \mathbf{Usp} \ n \quad as : \mathbf{Tel} \ n}{\mathbf{Els}p \ a \ as : \star}$$

with all the same constructors as  $\mathbf{Ucf}$  and  $\mathbf{Elcf}$  and additionally a constructor for constant exponentiation:

$$\frac{A : \star \quad b : \mathbf{Usp} \ n}{\text{'arr'} \ A \ b : \mathbf{Usp} \ n}$$

and a corresponding constructor for  $\mathbf{Els}p$ :

$$\frac{f : A \rightarrow \mathbf{Els}p \ b \ as}{\text{fun } f : \mathbf{Els}p \ (\text{'arr'} \ A \ b) \ as}$$

It is now easy to represent ordinals in this universe:

$$\underline{\text{let}} \frac{}{\mathbf{ord} : \mathbf{Usp} \ n} \quad \mathbf{ord} \Rightarrow \text{'mu'} \ (\text{'plus'} \ \text{'1'} \ (\text{'plus'} \ \text{vl} \ (\text{'arr'} \ \mathbf{Nat} \ \text{vl})))$$

We have been cheating a bit, because the constructor  $\text{'arr'}$  refers to a type. Thus  $\mathbf{Usp} \ n : \star_{i+1}$  if  $A : \star_i$  in  $\text{'arr'} \ A \ b$ . An alternative would be to insist that the codomain of  $\text{'arr'}$  is a closed strictly positive type, but this causes problems when introducing  $\text{fun}$  because of a negative occurrence of  $\mathbf{Els}p$ .

*Exercise 8.* Derive the constructors for  $\mathbf{ord} : \mathbf{Usp} \ n$  and a view which allows pattern matching over  $\mathbf{ord}$ . Use this to define ordinal addition for the encoded ordinals.

#### 4.4 Generic map

We don't know many useful generic operations which apply to closed strictly positive types, but there is an important one for open ones: generic map. While we have given only an interpretation for closed types we are able to express generic map by introducing maps between telescopes.

We introduce a family representing maps between telescopes, which correspond to a sequence of maps between the components of the telescopes:

$$\underline{\text{data}} \frac{as, bs : \mathbf{Tel} \ n}{\mathbf{Map} \ as \ bs : \star}$$



and generic map simply lifts maps on telescope to a function between the element of a type instantiated with the telescopes:

$$\text{let } \frac{fs : \text{Map } as \ bs \quad x : \text{Elspl } a \ as}{\text{gmap } fs \ x : \text{Elspl } a \ bs}$$

What are the constructors for **Map**? There are two obvious ones, which correspond to the idea that **Map** *as* *bs* is simply a sequences of maps between the components of *as* and *bs*:

$$\frac{}{\text{mnil} : \text{Map } \text{tnil} \ \text{tnil}} \quad \frac{f : \text{Elspl } a \ as \rightarrow \text{Elspl } b \ bs \quad fs : \text{Map } as \ bs}{\text{mcons } f \ fs : \text{Map } (\text{tcons } a \ as) \ (\text{tcons } b \ bs)}$$

However, it is useful to introduce a 3rd constructor, which extends a given sequence of maps by the identity function:

$$\frac{fs : \text{Map } as \ bs}{\text{mext } fs : \text{Map } (\text{tcons } a \ as) \ (\text{tcons } a \ bs)}$$

Note that **mext** *fs* isn't just **mcons** ( $\lambda x \Rightarrow x$ ) *fs* because instead of the identity we need a function of the type **Elspl** *a as*  $\rightarrow$  **Elspl** *a bs*. It would be possible to define **mext** mutually with **gmap** but it is much easier to introduce an additional constructor which also keeps the program structural recursive.

The definition of **gmap** is now rather straightforward by structural recursion on the argument:

```

gmap fs x  $\Leftarrow$  rec x
gmap fs x  $\Leftarrow$  case x
gmap fs (inl x)  $\Rightarrow$  inl (gmap fs x)
gmap fs (inr y)  $\Rightarrow$  inr (gmap fs y)
gmap fs void  $\Rightarrow$  void
gmap fs (pair x y)  $\Rightarrow$  pair (gmap fs x) (gmap fs y)
gmap fs (fun f)  $\Rightarrow$  fun ( $\lambda x \Rightarrow$  gmap fs (f x))
gmap fs (top x)  $\Leftarrow$  case fs
gmap (mcons f fs) (top x)  $\Rightarrow$  top (f x)
gmap (mext fs) (top x)  $\Rightarrow$  top (gmap fs x)
gmap fs (pop x)  $\Leftarrow$  case fs
gmap (mcons f fs) (pop x)  $\Rightarrow$  pop (gmap fs x)
gmap (mext fs) (pop x)  $\Rightarrow$  pop (gmap fs x)
gmap fs (push x)  $\Rightarrow$  push (gmap (mext fs) x)
gmap fs (in x)  $\Rightarrow$  in (gmap (mext fs) x)

```

The cases for the proper data constructors (**inl**, **inr**, **void**, **pair**, **fun** and **in**) are standard and just push **gmap** under the constructor. The other cases deal with the environment: **top** and **pop** have to analyse whether the environment has been constructed using **mcons** or **mext** while in the line for **push** we can reap the fruits by using **mext** instead having to use a non-structural recursive call to **gmap**.

As before in the case of **geq** the program is data-driven, i.e. we never have to inspect the type. However, the type-discipline helps us to find the right definition, which in many cases is the only possible one.

*Exercise 9.* Instantiate, **gmap** for **list**:

$$\text{let } \frac{}{\mathbf{list} : \mathbf{Usp}(\mathbf{suc} \ n)} \quad \mathbf{list} \Rightarrow \text{'mu'} \ (\text{'plus'} \ \text{'1'} \ (\text{'times'} \ (\mathbf{wk} \ \mathbf{vl}) \ \mathbf{vl}))$$

to obtain

$$\text{let } \frac{f : \mathbf{Els}p \ a \ as \rightarrow \mathbf{Els}p \ b \ as \quad xs : \mathbf{Els}p \ \mathbf{list} \ (\mathbf{tcons} \ a \ as)}{\mathbf{map} \ f \ xs : \mathbf{Els}p \ \mathbf{list} \ (\mathbf{tcons} \ b \ as)}$$

## 4.5 Relating universes

In the previous section we have incrementally defined three universes, each one extending the previous one together with a *typical* generic operation:

	universe of	inhabited by	generic operation
<b>Ufin</b>	finite types	Booleans ( <b>bool</b> )	Enumeration ( <b>enum</b> )
<b>Ucf</b>	context-free types	Rose trees ( <b>rt</b> )	Equality ( <b>geq</b> )
<b>Usp</b>	strictly positive types	Ordinals ( <b>ord</b> )	Map ( <b>gmap</b> )

We could have factored out the common parts of the definitions and established that every universe can be embedded into the next one but for pedagogical reasons we chose the incremental style of presentation. We note that the generic operations are typical for a given universe because they do not extend to the next level, i.e. enumeration doesn't work for context-free types because they contain types with an infinite number of elements; equality doesn't work for strictly positive types because equality here is in general undecidable (e.g. for ordinals).

Are there any important universes we have left out? Between **Ufin** and **Ucf** we can find the universe of regular types, i.e. types which are represented as regular expressions, where the datatype of lists plays the role of Kleene's star. Another possibility is to also allow coinductive context-free types like streams by including codes for terminal coalgebras, e.g.  $\mathbf{Stream} \ X = \nu X. A \times X$ . However, this doesn't fit very well with our way to define **El** inductively.

What about the universe of positive types extending the strictly positive types? It is unclear how to understand a type like  $\mu X. (X \rightarrow \mathbf{Bool}) \rightarrow \mathbf{Bool}$ <sup>8</sup> intuitively and there seem to be only very limited applications of positive inductive types. However, for **gmap** it is sensible to allow parameters in non-strict positive positions without closing under  $\mu$ .

<sup>8</sup> Note that  $(X \rightarrow \mathbf{Bool}) \rightarrow \mathbf{Bool}$  is covariant, unlike  $X \rightarrow \mathbf{Bool} \times \mathbf{Bool}$  which is contravariant.

## 4.6 Universes and Representation Types

There is a very strong connection between the notion of universe in Type Theory and the more recent notion of *representation type* which has emerged from work on type analysis [17] to become a popular basis for generic programming in Haskell [23, 22, 40]. The two notions are both standard ways to give a data representation to a collection of things, in this case, types:

- Martin-Löf’s universes  $(U, El)$  collect types as the *image of a function*  $El : U \rightarrow \star$ . Elements of  $U$  may thus be treated as proxies for the types to which they map.
- Representation types characterise a collection of types as a *predicate*,  $Rep : \star \rightarrow \star$ . An element of  $Rep T$  is a *proof* that  $T$  is in the collection, and it is also a piece of data from which one may compute.

The former approach is not possible in Haskell, because  $U \rightarrow \star$  is not expressible when  $U$  is a type rather than a ‘kind’. However, the latter has become possible, thanks to the recent extension of (ghc) Haskell with a type-indexed variant of inductive families [20], the so-called ‘Generalized Algebraic Data Types’ [38]. For example, one might define a universe of regular expression types as follows

```
data Rep a where
  Char    :: Rep Char
  Unit    :: Rep ()
  Pair    :: Rep a -> Rep b -> Rep (a,b)
  Either  :: Rep a -> Rep b -> Rep (Either a b)
  List    :: Rep a -> Rep [a]
```

and then write a function generic with respect to this universe by pattern-matching on  $Rep$ , always making sure to keep the the type representative to the left of the data which indirectly depends on it:

```
string :: Rep a -> a -> String
string Char    c      = [c]
string Unit    ()     = ""
string (Pair a b) (x, y) = string a x ++ string b y
string (Either a b) (Left x) = string x
string (Either a b) (Right y) = string y
string (List a)  xs     = xs >>= string a
```

Of course, in Epigram, these two kinds of collection are readily interchangeable. Given  $U$  and  $El$ , we may readily construct the predicate for ‘being in the image of  $El$ ’:

$$Rep X \Rightarrow Sigma U (\lambda u \Rightarrow El u = X)$$

In the other direction, given some  $Rep$ , we may ‘name’ a type as the dependent pair of the type itself and its representation: we interpret such a pair by

projecting out the type!

$$\begin{aligned} \mathbf{U} &\Rightarrow \mathbf{Sigma} \star (\lambda X \Rightarrow \mathbf{Rep} X) \\ \mathbf{El} &\Rightarrow \mathbf{fst} \end{aligned}$$

One can simulate this to some extent in Haskell by means of *existential* types

```
data U = forall a. U (Rep a)
```

and thus provide a means to compute one type from another—some sort of auxiliary data structure, perhaps—by writing a function `aux :: U -> U`. For example, `aux` might compute the notion of ‘one-hole context’ appropriate to its argument, in an attempt to support generic structure editing. Unfortunately, `U` is not a `Sigma` type but a System F ‘weak’ existential: direct access to the type it packages is not possible. There is no way to express operations whose types explicitly invoke these functions, e.g., plugging a value into a one-hole context to recover an element of the original type.

System F’s capacity for secrecy is rather useful in other circumstances, but it is a problem here. The nearest we can get to `El` is a rank-2 accessor which grants *temporary* access to the witness to facilitate a computation whose type does not depend on it.

```
for :: U -> (forall a. Rep a -> b) -> b
```

This problem is not the fault of representation types as opposed to universes (although the latter are a little neater for such tasks): it’s just a shortfall in the expressivity of Haskell.

## 5 Containers

In section 3 we started with a semantic definition of the universe of finite types, while in the previous section we introduced universes syntactically, i.e. using inductive definitions. In the present section we will exploit our work on container types to give a semantic interpretation of the universe of context-free types which also works for strictly positive types. It is good to have both views of the universes available, we have seen that the inductive approach is very practical to define generic and non-generic operations on data. However, the semantic approach we introduce here often provides an alternative approach to defining generic functions semantically. We will demonstrate this in more detail in the next section using the example of derivatives of datatypes. Another advantage of the semantic view is that it allows us to interpret open datatypes directly as operations on types, e.g. we can apply `list` to types which don’t have a name in our universe.

## 5.1 Unary containers

Before embarking on the more general concept of  $n$ -ary containers, which as we will see can model exactly the universe of strictly positive types, we have a look at unary containers, which model type constructors with one parameter, i.e. inhabitants of  $\star \rightarrow \star$ , where the parameter represents a type of *payload* elements to be stored within some structure. A unary container is given by a type of shapes  $S : \star$  and a family of positions  $P : S \rightarrow \star$ . I.e. we define:

$$\begin{array}{l} \text{data } \overline{\text{UCont}} : \star \\ \text{where } \frac{S : \star \quad P : S \rightarrow \star}{\text{ucont } S P : \text{UCont}} \end{array} \quad \begin{array}{c} \rightarrow \\ \triangleleft \\ s : S \quad P s \\ \text{ucont } S P \end{array}$$

We illustrate containers with *triangle diagrams*, intended to resemble a node in a tree with its root at the left. As with **W**-types, we indicate a choice of shape in the apex, and the base then represents a *set* of points, dependent on the shape; the arrow at the top of the base indicates that the points in this set are positions for payload.

The *extension* of a container is the parametric datatype which it describes: its values consist of a choice of shape and an assignment of payload to positions, represented functionally.

$$\begin{array}{l} \text{data } \frac{C : \text{UCont} \quad X : \star}{\text{UExt } C X : \star} \\ \text{where } \frac{s : S \quad f : P s \rightarrow X}{\text{uext } s f : \text{UExt } (\text{ucont } S P) X} \end{array} \quad \begin{array}{c} \text{uext} \\ \triangleleft \\ s \quad f \bullet p \mapsto x \end{array}$$

We can also illustrate inhabitants of such a type by a diagram, labelling the base with the function  $f$  which takes each position  $p$  to some payload value  $x$ .


An example of a unary container is this representation of **List**:

$$\text{let } \overline{\text{cList}} : \text{UCont} \quad \text{cList} \Rightarrow \text{ucont Nat Fin} \quad \begin{array}{c} \rightarrow \\ \triangleleft \\ n : \text{Nat} \\ \vdots \\ \bullet \text{fs}^{n-1} \text{ fz} \\ \text{cList} \end{array}$$

The shape of a list is its length, i.e. a natural number, and a list with shape  $n : \text{Nat}$  has **Fin**  $n$  positions. We can re-implement the constructors for lists to target the container representation

$$\begin{array}{l} \text{let } \frac{i : \text{Fin zero}}{\text{noFin } i : X} \quad \text{noFin } i \leftarrow \text{case } i \\ \text{let } \overline{\text{cnil}} : \text{UExt cList } X \quad \text{cnil} \Rightarrow \text{uext zero noFin} \end{array} \quad \begin{array}{c} \text{uext} \\ \triangleleft \\ \text{zero} \end{array}$$

$$\text{let } \frac{x : X \quad f : \text{Fin } n \rightarrow X \quad i : \text{Fin } (\text{suc } n)}{\text{caseFin } x f i : X}$$

$$\begin{aligned} \text{caseFin } x f i &\Leftarrow \text{case } i \\ \text{caseFin } x f \text{fz} &\Rightarrow x \\ \text{caseFin } x f (\text{fs } i) &\Rightarrow f i \end{aligned}$$


$$\text{let } \frac{x : X \quad xs : \text{UExt cList } X}{\text{ccons } x xs : \text{UExt cList } X}$$

$$\begin{aligned} \text{ccons } x xs &\Leftarrow \text{case } xs \\ \text{ccons } x (\text{uext } n f) &\Rightarrow \text{uext } (\text{suc } n) (\text{caseFin } x f) \end{aligned}$$

*Exercise 10.* We can also give a container representation for binary trees, here shapes are given by trees containing data, positions by paths through such a tree:

$$\text{data } \frac{}{\text{cTreeS} : \star} \quad \text{where } \frac{}{\text{sleaf} : \text{cTreeS}} \quad \frac{l, r : \text{cTreeS}}{\text{snode } l r : \text{cTreeS}}$$

$$\text{data } \frac{s : \text{cTreeS}}{\text{cTreeP } s : \text{Type}} \quad \text{where } \frac{}{\text{phere} : \text{cTreeP } (\text{node } l r)}$$

$$\frac{q : \text{cTreeP } l}{\text{pleft } q : \text{cTreeP } (\text{node } l r)} \quad \frac{p : \text{cTreeP } r}{\text{pright } p : \text{cTreeP } (\text{node } l r)}$$

$$\text{let } \frac{}{\text{cTree} : \text{UCont}} \quad \text{cTree} \Rightarrow \text{uext cTreeS cTreeP}$$

Implement `leaf` and `node` for `cTree`:

$$\text{let } \frac{}{\text{cleaf} : \text{UExt cTree } X}$$

$$\text{let } \frac{r : \text{UExt cTree } X \quad x : X \quad r : \text{UExt cTree } X}{\text{cnode } l x r : \text{UExt cTree } X}$$

Each container gives rise to a functor. We can implement `map` for unary containers by applying the function to be mapped directly to the payload:

$$\text{let } \frac{C : \text{UCont} \quad f : X \rightarrow Y \quad c : \text{UExt } C X}{\text{ucmap } C f c : \text{UExt } C Y}$$

$$\begin{aligned} \text{ucmap } C f c &\Leftarrow \text{case } c \\ \text{ucmap } (\text{ucont } S P) f (\text{uext } s g) &\Rightarrow \text{uext } s (\lambda x \Rightarrow f (g x)) \end{aligned}$$

A morphism between functors is a natural transformation, e.g. `reverse` is a natural transformation from list to list. We can explicitly represent morphisms between containers: given unary containers `ucont S P` and `ucont T Q`, a morphism is a function on shapes  $f : S \rightarrow T$  and a family of functions on positions, which assigns to every position in the target a position in the source, i.e.


$$u : \forall s : S \Rightarrow Q (f s) \rightarrow P s$$

The contravariance of the function on positions may be surprising, however, it can be intuitively understood by the fact that we can always say where a piece of payload comes from but not where it goes to, since it may be copied or disappear. Hence we define:

$$\text{data } \frac{C, D : \mathbf{UCont}}{\mathbf{UMor } C D : \star} \quad \text{where } \frac{f : S \rightarrow T \quad u : \forall s : S \Rightarrow Q (f s) \rightarrow P s}{\mathbf{umor } f u : \mathbf{UMor} (\mathbf{ucont } S P) (\mathbf{ucont } T Q)}$$

To every formal morphism between containers we assign a family of maps, parametric in the payload type:

$$\begin{aligned} \text{let } & \frac{m : \mathbf{UMor } C D \quad c : \mathbf{UExt } C X}{\mathbf{UMapp } m c : \mathbf{UExt } D X} \\ & \mathbf{UMapp } m c \Leftarrow \text{case } m \\ & \mathbf{UMapp } (\mathbf{umor } f u) c \Leftarrow \text{case } c \\ & \mathbf{UMapp } (\mathbf{umor } f u) (\mathbf{uext } s g) \Rightarrow \\ & \quad \mathbf{uext } (f s) (\lambda q \Rightarrow g (u s q)) \end{aligned}$$



As an example we can define **cHead** for the container representation of lists, since we require totality we will define a morphism between **cList** and **cMaybe**, which relates to Haskell's **Maybe** type:

$$\text{let } \frac{}{\mathbf{cMaybe} : \mathbf{UCont}} \quad \mathbf{cMaybe} \Rightarrow \mathbf{ucont } \mathbf{Bool } \mathbf{So}$$

There are two possible layouts for **cMaybe** containers:



There are then two cases to consider when defining our morphism. For the **zero** shape of **cList**, we choose the **false** shape of **cMaybe**, leaving no positions to fill. For any other input shape, we choose **true**, leaving one position to fill: we fetch its payload from input position **fz**—the head.

$$\text{let } \frac{n : \mathbf{Nat}}{\mathbf{isSuc} : \mathbf{Bool}}$$

$$\begin{aligned} \mathbf{isSuc } n & \Leftarrow \text{case } n \\ \mathbf{isSuc } \mathbf{zero} & \Rightarrow \mathbf{false} \\ \mathbf{isSuc } (\mathbf{suc } n) & \Rightarrow \mathbf{true} \end{aligned}$$


$$\text{let } \frac{n : \mathbf{Nat} \quad q : \mathbf{So} (\mathbf{isSuc } n)}{\mathbf{least } n q : \mathbf{Fin } n}$$

$$\begin{aligned} \mathbf{least } n q & \Leftarrow \text{case } n \\ \mathbf{least } \mathbf{zero } q & \Leftarrow \text{case } q \\ \mathbf{least } (\mathbf{suc } n) q & \Rightarrow \mathbf{fz} \end{aligned}$$

$$\text{let } \frac{}{\mathbf{cHead} : \mathbf{UMor } \mathbf{cList } \mathbf{cMaybe}}$$

$$\mathbf{cHead} \Rightarrow \mathbf{umor } \mathbf{isSuc } \mathbf{least}$$

We illustrate these two cases as follows:



It is not hard to show that these families of maps are always natural transformations in the categorical sense, with respect to `UExt`'s interpretation of unary containers as functors. Indeed, it turns out that all natural transformations between functors arising from containers can be given as container morphisms, see theorem 3.4. in [2].

*Exercise 11.* Give the representation of `reverse` as morphism between unary containers, i.e.

$$\text{let } \overline{\mathbf{cRev}} : \mathbf{UMor} \mathbf{cList} \mathbf{cList}$$

*Exercise 12.* While the interpretation of morphisms is full, i.e. every natural transformation comes from a container morphism, the same is not true for containers as representations of functors. Can you find a functor which is not representable as a unary container?

## 5.2 *n*-ary containers

We are now going to interpret strictly positive types `Usp` as containers by implementing operations on containers which correspond to constructors of `Usp`. We reap the fruits by defining a simple evaluation function which `evalC` which interprets `Usp`s as containers. First of all we have to generalize our previous definition to *n*-ary containers to reflect the presence of variables in `Usp`:

$$\begin{array}{l} \text{data } \frac{n : \mathbf{Nat}}{\mathbf{Cont} \ n : \star} \\ \text{where } \frac{S : \star \quad P : \mathbf{Fin} \ n \rightarrow S \rightarrow \star}{\mathbf{cont} \ S \ P : \mathbf{Cont} \ n} \end{array} \quad \begin{array}{c} \begin{array}{c} \rightarrow \mathbf{fz} \\ P \ \mathbf{fz} \ s \\ \vdots \\ \rightarrow \mathbf{fs}^{n-1} \ \mathbf{fz} \\ P \ (\mathbf{fs}^{n-1} \ \mathbf{fz}) \ s \end{array} \\ \begin{array}{c} s : S \\ \mathbf{cont} \ S \ P \end{array} \end{array}$$

It is important to understand that we use only one shape but *n* sets of positions. E.g. consider the two-parameter container of leaf and node labelled trees, the shape of a tree is given by ignoring the data, but the positions for leaf-data and tree-data are different. Accordingly, in our diagrams, we may segment the base



of the triangle to separate the surfaces where each sort of payload attaches and we index the arrows accordingly.

The extension of an  $n$ -ary container is given by an operator on a sequence of types, generalizing the sketch above to the  $n$ -ary case:

$$\begin{array}{l}
 \text{data } \frac{C : \text{Cont } n \quad X_s : \text{Fin } n \rightarrow \star}{\text{Ext } C \ X_s : \star} \\
 \\
 \text{where } \frac{P : \text{Fin } n \rightarrow S \rightarrow \star \quad X_s : \text{Fin } n \rightarrow \star}{s : S \quad f : \forall i : \text{Fin } n \Rightarrow P \ i \ s \rightarrow X_s \ i} \\
 \frac{\text{ext } s \ f : \text{Ext } (\text{cont } S \ P) \ X_s}{}
 \end{array}$$

*Exercise 13.* Show that  $n$ -ary containers give rise to  $n$ -ary functors, i.e. implement:

$$\text{let } \frac{C : \text{Cont } n \quad X_s, Y_s : \text{Fin } n \rightarrow \star \quad fs : \forall i : \text{Fin } n \Rightarrow X_s \ i \rightarrow Y_s \ i \quad x : \text{Ext } C \ X_s}{\text{map } C \ fs \ x : \text{Ext } C \ Y_s}$$

### 5.3 Coproducts and products

A constant operator is represented by a container which has no positions, e.g. the following containers represent the empty and the unit type:

$$\text{let } \frac{}{\mathbf{cZero} : \text{Cont } n} \quad \mathbf{cZero} \Rightarrow \text{cont } \mathbf{Zero} \ (\lambda i; s \Rightarrow \mathbf{Zero})$$

$$\text{let } \frac{}{\mathbf{cOne} : \text{Cont } n} \quad \mathbf{cOne} \Rightarrow \text{cont } \mathbf{One} \ (\lambda i; s \Rightarrow \mathbf{Zero})$$



Given two containers  $C = \text{cont } S \ P, D = \text{cont } T \ Q$  we construct their coproduct or sum, representing a choice between  $C$  and  $D$ . On the shapes this is just the type-theoretic coproduct  $\text{Plus } S \ T$  as defined earlier. What is a position in  $\text{Plus } S \ T$ ? If our shape is of the form  $\text{Inl } s$  then it is given by  $P \ s$ , on the other hand if it is of the form  $\text{Inr } t$  then it is given by  $Q \ t$ . Abstracting shapes and positions, we arrive at:

$$\text{data } \frac{P : A \rightarrow \star \quad Q : B \rightarrow \star \quad ab : \text{Plus } A \ B}{\text{PPlus } P \ Q \ ab : \star}$$

$$\text{where } \frac{p : P \ a}{\text{pinl } p : \text{PPlus } P \ Q \ (\text{Inl } a)} \quad \frac{q : Q \ b}{\text{pinr } q : \text{PPlus } P \ Q \ (\text{Inr } b)}$$

Putting everything together we define the containers as follows, with the two typical layouts shown in the diagrams:

$$\begin{array}{l} \text{let } \frac{C, D : \text{Cont } n}{\mathbf{cPlus } C D : \text{Cont } n} \\ \mathbf{cPlus } C D \leftarrow \text{case } C \\ \mathbf{cPlus } (\text{cont } S P) D \leftarrow \text{case } D \\ \mathbf{cPlus } (\text{cont } S P) (\text{cont } T Q) \\ \Rightarrow \text{cont } (\text{Plus } S T) \\ (\lambda i \Rightarrow \text{PPlus } (P i) (Q i)) \end{array} \quad \begin{array}{l} \begin{array}{c} \rightarrow_i \\ \text{Inl } s \\ \bullet \text{pinl } (p : P i s) \end{array} \\ \mathbf{cPlus } C D \\ \begin{array}{c} \rightarrow_i \\ \text{Inr } t \\ \bullet \text{pinr } (q : Q i t) \end{array} \\ \mathbf{cPlus } C D \end{array}$$

Let's turn our attention to products: on shapes again this is just the type-theoretic product, **Times**—each component has a shape. Given two containers  $C = \text{cont } S P, D = \text{cont } T Q$ , as above, what are the positions in a product shape  $\text{Pair } s t : \text{Times } S T$ ? There are two possibilities: either the position is in the left component, then it is given by  $P s$  or it is in the right component then it is given by  $Q t$ . Abstracting shapes and positions again we define abstractly:

$$\begin{array}{l} \text{data } \frac{P : A \rightarrow \star \quad Q : B \rightarrow \star \quad ab : \text{Times } A B}{\text{PTimes } P Q ab : \star} \text{ where} \\ \frac{p : P a}{\text{pleft } p : \text{PTimes } P Q (\text{Pair } a b)} \quad \frac{q : Q b}{\text{pright } q : \text{PTimes } P Q (\text{Pair } a b)} \end{array}$$

and we define the product of containers as:

$$\begin{array}{l} \text{let } \frac{C, D : \text{Cont } n}{\mathbf{cTimes } C D : \text{Cont } n} \\ \mathbf{cTimes } C D \leftarrow \text{case } C \\ \mathbf{cTimes } (\text{cont } S P) D \leftarrow \text{case } D \\ \mathbf{cTimes } (\text{cont } S P) (\text{cont } T Q) \\ \Rightarrow \text{cont } (\text{Times } S T) \\ (\lambda i \Rightarrow \text{PTimes } (P i) (Q i)) \end{array} \quad \begin{array}{l} \begin{array}{c} \rightarrow_i \\ s \\ \bullet \text{pleft } (p : P i s) \end{array} \\ \text{Pair} \\ \begin{array}{c} \rightarrow_i \\ t \\ \bullet \text{pright } (q : Q i t) \end{array} \\ \mathbf{cTimes } C D \end{array}$$


*Exercise 14.* Define an operation on containers which interprets constant exponentiation as described in section 4.3, i.e. define

$$\text{let } \frac{A : \star \quad C : \text{Cont } n}{\mathbf{cArr } A C : \text{Cont } n}$$

## 5.4 Structural operations

If we want to interpret the universe of context-free or strictly positive types faithfully, we also have to find counterparts for the structural operation **vl** (last variable), **wk** (weakening) and **def** (local definition).

The interpretation of **vl** is straightforward: There is only one shape and in the family of positions  $P : \mathbf{Fin}(\mathbf{succ} n)$  there is only one position at index **fz**:

$$\text{let } \frac{}{\mathbf{cvl} : \mathbf{Cont}(\mathbf{succ} n)} \quad \mathbf{cvl} \Rightarrow \mathbf{cont} \mathbf{One}(\lambda i; s \Rightarrow i = \mathbf{fz})$$


Weakening isn't much harder: the shape stays the same but the position indices get shifted by one assigning no positions to index **fz**. We define first an auxiliary operator on positions:

$$\text{let } \frac{P : \mathbf{Fin} n \rightarrow S \rightarrow \star \quad i : \mathbf{Fin}(\mathbf{succ} n) \quad s : S}{\mathbf{Pwk} P i s : \star}$$

$$\mathbf{Pwk} P i s \Leftarrow \text{case } i$$


$$\mathbf{Pwk} P \mathbf{fz} s \Rightarrow \mathbf{Zero}$$

$$\mathbf{Pwk} P (\mathbf{fs} i) s \Rightarrow P i s$$


and use this to define:

$$\text{let } \frac{C : \mathbf{Cont} n}{\mathbf{cwk} C : \mathbf{Cont}(\mathbf{succ} n)}$$

$$\mathbf{cwk} C \Leftarrow \text{case } C$$

$$\mathbf{cwk}(\mathbf{cont} S P) \Rightarrow \mathbf{cont} S(\mathbf{Pwk} P)$$


The case of local definition is more interesting. We assume as given two containers:  $C = \mathbf{cont} S P : \mathbf{Cont}(\mathbf{succ} n)$ ,  $D = \mathbf{cont} T Q : \mathbf{Cont} n$ . We create a new  $n$ -ary container by *binding* variable **fz** of  $C$  to  $D$ , hence attaching  $D$ -structures to each **fz**-indexed position of a  $C$ -structure. The  $i$ -positions of the result correspond either to  $i$ -positions of some inner  $D$ , or the *free* (**fs** $i$ )-positions of the outer  $C$ .



To record the shape of the whole thing, we need to store the outer  $C$  shape, some  $s : S$ , and the inner  $D$  shapes: there is one for each outer **fz**-position, hence we need a function  $f : P \mathbf{fz} s \rightarrow T$ . As before we abstract from the specific position types and define abstractly:

$$\text{data } \frac{S : \star \quad P_0 : S \rightarrow \star \quad T : \star}{\mathbf{Sdef} S P_0 T : \star} \quad \text{where } \frac{s : S \quad f : P_0 s \rightarrow T}{\mathbf{sdef} s f : \mathbf{Sdef} S P_0 T}$$

What is a position in the new container, for a given index  $i$ ? It must either be a 'free' outer position, given by  $P(\mathbf{fs} i)$ , or the pair of a 'bound' outer position

with an inner position given by  $Q\ i$ . Hence, we define a general operator for positions in **Sdef**, which we can instantiate suitably for each index:

$$\text{data } \frac{P_0, P' : S \rightarrow \star \quad Q : T \rightarrow \star \quad x : \text{Sdef } S P_0 T}{\text{Pdef } P_0 P' Q x : \star}$$

$$\text{where } \frac{p : P' s}{\text{ppos } p : \text{Pdef } P_0 P' Q (\text{sdef } s f)} \quad \frac{p : P_0 s \quad q : Q (f p)}{\text{qpos } p q : \text{Pdef } P_0 P' Q (\text{sdef } s f)}$$


Putting the components together, we can present the definition operator:

$$\text{let } \frac{C : \text{Cont } (\text{suc } n) \quad D : \text{Cont } n}{\text{cdef } C D : \text{Cont } n}$$

$$\begin{aligned} \text{cdef } C D &\Leftarrow \text{case } C \\ \text{cdef } (\text{cont } S P) D &\Leftarrow \text{case } D \\ \text{cdef } (\text{cont } S P) (\text{cont } T Q) &\Rightarrow \text{cont } (\text{Sdef } S (P \text{fz}) T) (\lambda i \Rightarrow \text{Pdef } (P \text{fz}) (P (\text{fs } i)) (Q i)) \end{aligned}$$

### 5.5 Inductive types ( $\mu$ )

To interpret the **mu** constructor we take an  $n + 1$ -ary container  $C = \text{cont } S P : \text{Cont } (\text{suc } n)$  and try to find a container which represents the initial algebra with respect to the ‘bound’ index **fz**. For each shape  $s : S$ ,  $P \text{fz } s$  gives the positions of recursive subobjects. Meanwhile the positions for  $i$ -indexed payload at each node are given by  $P (\text{fs } i)$ .



Clearly, to be able to construct a tree at all, there must be at least one ‘base case’  $s$  for which  $P \text{fz } s$  is empty. Otherwise there are no leaves and the corresponding tree type is empty.

How can we describe the shapes of these trees? At each node, we must supply the top-level shape, together with a function which gives the shape of the subtrees. This is given exactly by  $\text{W } S (P \text{fz})$ . Given a shape in form of a **W**-tree, the positions at index  $i$  correspond to path leading to a  $P (\text{fs } i)$  somewhere in the tree. We can define the types of paths in a tree in general:

$$\text{data } \frac{S : \star \quad P_0, P' : S \rightarrow \star \quad x : \text{W } S P_0}{\text{PW } S P_0 P' x : \star}$$

$$\text{where } \frac{x : P' s}{\text{here } x : \text{PW } S P_0 P' (\text{Sup } s f)} \quad \frac{p : P_0 s \quad r : \text{PW } S P_0 P' (f p)}{\text{under } p r : \text{PW } S P_0 P' (\text{Sup } s f)}$$

The idea is that a path either exits at the top level node **here** at a position in  $P' s$  or continues into the subtree **under** a positions in  $P_0 s$ . Putting shapes and paths together we arrive at the following definition:

$$\begin{aligned} \underline{\text{let}} \quad & \frac{C : \text{Cont}(\text{suc } n)}{\text{cMu } C : \text{Cont } n} \\ \text{cMu } C & \leftarrow \underline{\text{case}} C \\ \text{cMu}(\text{cont } S P) & \Rightarrow \text{cont}(W S (P \text{fz})) (\lambda i \Rightarrow \text{PW } S (P \text{fz}) (P (\text{fs } i))) \end{aligned}$$

## 5.6 Interpreting universes

Since we have constructed semantic counterparts to every syntactic constructor in **Ucf** we can interpret any type name by a container with the corresponding arity:

$$\begin{aligned} \underline{\text{let}} \quad & \frac{a : \text{Ucf } n}{\text{evalC } a : \text{Cont } n} \\ \text{evalC } a & \leftarrow \underline{\text{rec}} a \\ \text{evalC } a & \leftarrow \underline{\text{case}} a \\ \text{evalC } \text{vl} & \Rightarrow \text{cvl} \\ \text{evalC } (\text{wk } a) & \Rightarrow \text{cwk}(\text{evalC } a) \\ \text{evalC } '0' & \Rightarrow \text{cZero} \\ \text{evalC } ('plus' a b) & \Rightarrow \text{cPlus}(\text{evalC } a)(\text{evalC } b) \\ \text{evalC } '1' & \Rightarrow \text{cOne} \\ \text{evalC } ('times' a b) & \Rightarrow \text{cTimes}(\text{evalC } a)(\text{evalC } b) \\ \text{evalC } (\text{def } f a) & \Rightarrow \text{cdef}(\text{evalC } f)(\text{evalC } a) \\ \text{evalC } ('mu' f) & \Rightarrow \text{cMu}(\text{evalC } f) \end{aligned}$$

Combining **evalC** with **Ext** we can assign to any name in **Ucf** an operator on types:

$$\underline{\text{let}} \quad \frac{a : \text{Ucf } n \quad Xs : \text{Fin } n \rightarrow \star}{\text{eval } a Xs : \star} \quad \text{eval } a Xs \Rightarrow \text{Ext}(\text{evalC } a) Xs$$

The advantage is that we can apply our operators to any types, not just those which have name. Using the solution to exercise 13 we also obtain a generic map function.

So far we have only interpreted the type names, i.e. the inhabitants of **Ucf**  $n$ , what about the elements, i.e. the inhabitants of **Elcf**  $aas$ ? Using **Ext** we can define

a semantic version of `Elcf`:

$$\text{data } \frac{n : \text{Nat}}{\text{CTel } n : \star} \text{ where } \frac{}{\text{ctnil} : \text{Tel zero}} \frac{a : \text{Cont } n \quad as : \text{Tel } n}{\text{ctcons } a \text{ } as : \text{Tel (suc } n)}$$

$$\text{let } \frac{Cs : \text{Tel } n \quad i : \text{Fin } n}{\text{TelEl } Cs \text{ } i : \star}$$

$$\text{TelEl } Cs \text{ } i \leftarrow \text{rec } Cs$$

$$\text{TelEl } Cs \text{ } i \leftarrow \text{case } i$$

$$\text{TelEl } Cs \text{ } fz \leftarrow \text{case } Cs$$

$$\text{TelEl (tcons } C \text{ } Cs) \text{ } fz \Rightarrow \text{Ext } C \text{ (TelEl } Cs)$$

$$\text{TelEl } Cs \text{ (fs } i) \leftarrow \text{case } Cs$$

$$\text{TelEl (tcons } C \text{ } Cs) \text{ (fs } i) \Rightarrow \text{TelEl } Cs \text{ } i$$

$$\text{let } \frac{C : \text{Cont } n \quad Cs : \text{Tel } n}{\text{CEl } C \text{ } Cs : \star} \quad \text{CEl } C \text{ } Cs \Rightarrow \text{Ext } C \text{ (TelEl } Cs)$$

*Exercise 15.* Implement semantic counterparts of the constructor for `Elcf` giving rise to an interpretation of `Elcf` by `CEl`. Indeed, this interpretation is exhaustive and disjoint.

## 5.7 Small containers

We have given a translation of the context-free types as containers, but as exercise 14 shows, these capture more than just the context-free types, in fact it corresponds to the strictly positive universe. As a result we cannot derive a semantic version of generic equality which is *typical* of the smaller universe.

We can, however, define a notion of container which captures precisely the context-free types and give a semantic version `geq` for these containers which we christen ‘small containers’.

A container is small if there is a decidable equality on its shapes and if the positions at a given shape are finite, so:

$$\text{let } \frac{A : \star}{\text{DecEq } A : \star}$$

$$\text{DecEq } A \Rightarrow \forall a, a' : A \Rightarrow \text{Dec } (a = a')$$

$$\text{data } \frac{n : \text{Nat}}{\text{SCont } n : \star}$$

$$\text{where } \frac{S : \star \quad eqS : \text{DecEq } S \quad P : \text{Fin } n \rightarrow S \rightarrow \text{Nat}}{\text{scont } S \text{ } eqS \text{ } P : \text{SCont } n}$$

$$\text{data } \frac{C : \text{SCont } Xs : \text{Fin } n \rightarrow \star}{\text{SExt } C \text{ } Xs : \star}$$

$$\text{where } \frac{s : S \quad f : \forall i : \text{Fin } n \Rightarrow \text{Fin } (P \text{ } i \text{ } s) \rightarrow Xs \text{ } i}{\text{sext } s \text{ } f : \text{SExt } C \text{ } Xs}$$

We can redefine the variable case, disjoint union, products and the fix point operator for these containers, for instance:

$$\begin{aligned} \text{let } & \frac{C, D : \mathbf{SCont } n}{\mathbf{SCTimes } C D : \mathbf{Cont } n} \\ \mathbf{SCTimes } C D & \leftarrow \text{case } C \\ \mathbf{SCTimes } (\text{scont } S \text{ eqS } P) D & \leftarrow \text{case } D \\ \mathbf{SCTimes } (\text{scont } S \text{ eqS } P) (\text{scont } T \text{ eqT } Q) & \\ \Rightarrow \text{scont } (\mathbf{Times } S T) & \\ & (\mathbf{TimesEq } \text{eqS } \text{eqT}) \\ & (\lambda i; s \Rightarrow \mathbf{plus } (P i s) (Q i s)) \end{aligned}$$

Where **TimesEq** is a proof that cartesian product preserves decidable equality by comparing pointwise:

$$\text{let } \frac{\text{eqS} : \mathbf{DecEq } S \quad \text{eqT} : \mathbf{DecEq } T}{\mathbf{TimesEq } \text{eqS } \text{eqT} : \mathbf{DecEq } (\mathbf{Times } S T)}$$

Our generic equality for small containers is then a proof that **SExt** preserves equality:

$$\text{let } \frac{C : \mathbf{SCont } n \quad Xs : \mathbf{Fin } n \rightarrow \star \quad \text{eqs} : \forall i : \mathbf{Fin } n \Rightarrow \mathbf{DecEq } (Xs i)}{\mathbf{SContEq } C Xs \text{eqs} : \mathbf{DecEq } (\mathbf{SExt } C Xs)}$$

*Exercise 16.* Complete the construction of **SCTimes** and develop operators constructing disjoint union, local definition, fixed points, and variables for small containers. Finally construct the definition of **SContEq**.

To work with Epigram's built in equality you will need to use the fact that application preserves equality:

$$\begin{aligned} \text{let } & \frac{f, g : S \rightarrow T \quad a, b : S \quad p : f = g \quad q : a = b}{\mathbf{applEq } p q : f a = g b} \\ \mathbf{applEq } p q & \leftarrow \text{case } p \\ \mathbf{applEq } \text{refl } q & \leftarrow \text{case } q \\ \mathbf{applEq } \text{refl } \text{refl} & \Rightarrow \text{refl} \end{aligned}$$

And that constructors are disjoint, so for example  $\mathbf{Inl } a = \mathbf{Inr } b$  is a provably empty type:

$$\text{let } \frac{a : A \quad b : B \quad p : (\mathbf{Inl } a : \mathbf{Plus } A B) = (\mathbf{Inr } b : \mathbf{Plus } A B)}{\mathbf{InlneqInr } p : X}$$

$$\mathbf{InlneqInr } p \leftarrow \text{case } p$$

## 6 Derivatives


In [24] Huet introduced the *zipper* as a datatype to represent a position within a tree. The basic idea is that at every step on the path to the current position, we

remember the context left over. E.g. in the example of unlabelled binary trees,

$$\text{data } \frac{}{\text{BT} : \star} \quad \text{where } \frac{}{\text{leaf} : \text{BT}} \quad \frac{l, r : \text{BT}}{\text{node } l r : \text{BT}}$$

the corresponding zipper type is:

$$\text{data } \frac{}{\text{Zipper} : \star}$$

$$\text{where } \frac{l : \text{Zipper} \quad r : \text{BT}}{\text{left } l r : \text{Zipper}} \quad \frac{l : \text{BT} \quad r : \text{Zipper}}{\text{right } l r : \text{Zipper}} \quad \frac{}{\text{here} : \text{Zipper}}$$


We can think of a **Zipper** as a tree with one subtree chopped out at the place marked **here**. One of the operations on a zipper is to plug a binary tree into the hole, i.e. we define:<sup>9</sup>


$$\text{let } \frac{z : \text{Zipper} \quad t : \text{BT}}{\text{plug } z t : \text{BT}}$$

$$\text{plug } z t \Leftarrow \text{rec } z$$

$$\text{plug } z t \Leftarrow \text{case } z$$

$$\text{plug } (\text{left } l r) t \Rightarrow \text{node } (\text{plug } l t) r$$

$$\text{plug } (\text{right } l r) t \Rightarrow \text{node } l (\text{plug } r t)$$

$$\text{plug } \text{here } t \Rightarrow t$$


Clearly, the zipper is a generic construction which should certainly work on any context-free type. When trying to express the general scheme of a zipper, Conor McBride realised that a zipper is always a sequence of basic steps which arise as the formal derivative of the functor defining the datatype. I.e. if our datatype is  $\mu X.F X$ , e.g.  $\mu X.1 + X \times X$  in the example of binary trees, then the corresponding zipper is  $\text{List}(\partial F(\mu X.F X))$ . In the binary tree example  $F X = 1 + X \times X$  and  $\partial F X = 2 \times X$ . Indeed **Zipper** is isomorphic to **List**  $(2 \times \text{BT})$ .

## 6.1 Derivatives of context-free types

We will here concentrate on the notion of the partial derivative of an  $n$ -ary operator on types, which corresponds to the type of *one hole contexts* of the given type. This is an alternative explanation of the formal laws of derivatives and we shall define an operator on context-free types following this intuition:

$$\text{let } \frac{a : \text{Ucf } n \quad i : \text{Fin } n}{\text{partial } a i : \text{Ucf } n}$$

<sup>9</sup> Here we have chosen the root-to-hole representation of contexts. Huet's hole-to-root presentation uses the same datatype. Both are useful, with the choice depending on where you need the most rapid access.




The parameter  $i$  denotes the argument on which we take the derivative, indeed the partial derivative really is a variable binding operation, this is obliterated by the usual notation  $\frac{\partial F}{\partial X}$  which really binds  $X$ .

We define this operation by structural recursion on  $a$ , let's consider the polynomial cases: what is the derivative, i.e. the type of one hole contexts of 'plus'  $a b$ ? We either have a hole in an element of  $a$  or a hole in an element of  $b$ , hence:

$$\mathbf{partial} ('plus' a b) i \Rightarrow 'plus' (\mathbf{partial} a i) (\mathbf{partial} b i)$$

Maybe slightly more interesting, what is the type of one-hole contexts of 'times'  $a b$ ? A hole in a pair is either a hole in the first component, leaving the second intact or symmetrically, a hole in the second, leaving the first intact. Hence we arrive at

$$\mathbf{partial} ('times' a b) i \Rightarrow 'plus' ('times' (\mathbf{partial} a i) b) ('times' a (\mathbf{partial} b i))$$



which indeed corresponds to the formal derivative of a product, although we arrived at it using a rather different explanation. Unsurprisingly, the derivative of a constant is '0', since there are no holes to plug:

$$\begin{aligned} \mathbf{partial} '0' i &\Rightarrow '0' \\ \mathbf{partial} '1' i &\Rightarrow '0' \end{aligned}$$

Structural operations like variables and weakening are usually ignored in Calculus, an omission we will have to fill here to be able to implement **partial** for those cases. In both cases we have to inspect  $i$ : for **vl** we have exactly one choice if  $i = \mathbf{fz}$  and none otherwise, hence we have:

$$\begin{aligned} \mathbf{partial} \mathbf{vl} \mathbf{fz} &\Rightarrow '1' \\ \mathbf{partial} \mathbf{vl} (\mathbf{fs} i) &\Rightarrow '0' \end{aligned}$$


In the case of **wk**  $a$  the situation is reversed, there is no choice if  $i = \mathbf{fz}$  and otherwise we recur structurally:

$$\begin{aligned} \mathbf{partial} (\mathbf{wk} a) \mathbf{fz} &\Rightarrow '0' \\ \mathbf{partial} (\mathbf{wk} a) (\mathbf{fs} i) &\Rightarrow \mathbf{wk} (\mathbf{partial} a i) \end{aligned}$$

The case of local definitions **def**  $f a$  corresponds to the chain rule in Calculus. An  $i$ - hole in an element of **def**  $f a$  is either a (**fs**  $i$ ) hole in the outer  $f$ , or it is

a hole in  $f$  for the defined variable  $fz$  together with an  $i$ -hole in some  $a$ . More formally we have:

$$\text{partial}(\text{def } f \ a) \ i \Rightarrow \text{'plus'}(\text{def}(\text{partial } f \ (\text{fs } i)) \ a) \ (\text{'times'}(\text{def}(\text{partial } f \ fz) \ a) \ (\text{partial } a \ i))$$




The case for initial algebras  $\text{'mu'} \ f$  has no counterpart in calculus. However, it can be derived using the chain rule above: we know that  $\text{'mu'} \ f$  is isomorphic to  $\text{def } f \ (\text{'mu'} \ f)$ . Now using the chain rule we arrive at

$$\text{'plus'}(\text{def}(\text{partial } f \ (\text{fs } i)) \ (\text{'mu'} \ f)) \ (\text{'times'}(\text{def}(\text{partial } f \ fz) \ (\text{'mu'} \ f)) \ (\text{partial}(\text{'mu'} \ f) \ i))$$

This expression is recursive in  $\text{partial}(\text{'mu'} \ f) \ i$  hence we obtain the formal derivative by taking the initial algebra of it, recording the contexts for a sequence of internal steps through the tree, terminated by the node with the external hole:

$$\text{partial}(\text{'mu'} \ f) \ i \Rightarrow \text{'mu'}(\text{'plus'}(\text{wk}(\text{def}(\text{partial } f \ (\text{fs } i)) \ (\text{'mu'} \ f))) \ (\text{'times'}(\text{wk}(\text{def}(\text{partial } f \ fz) \ (\text{'mu'} \ f)) \ \text{vl})))$$



A closer analysis shows that the use of initial algebras here is justified by the fact that we are only interested in holes which appear at some finite depths.

As an example consider the derivative of lists  $\text{partial list } fz$ : after applying some simplification we obtain  $\text{'mu'}(\text{'plus'}(\text{wklist})(\text{'times'} \ \text{vl}(\text{wkvl})))$  or reexpressed in a more standard notation  $\mu X.(\text{list } A) + A \times X$ , which can be easily seen to correspond to lists with a hole for  $A$ .

We summarize the definition of **partial**:

```

partial  $a\ i \Leftarrow \underline{\text{rec}}\ a$ 
partial  $a\ i \Leftarrow \underline{\text{case}}\ a$ 
partial  $\text{vl}\ i \Leftarrow \underline{\text{case}}\ i$ 
  partial  $\text{vl}\ \text{fz} \Rightarrow \text{'1'}$ 
  partial  $\text{vl}\ (\text{fs}\ i) \Rightarrow \text{'0'}$ 
partial  $(\text{wk}\ a)\ i \Leftarrow \underline{\text{case}}\ i$ 
  partial  $(\text{wk}\ a)\ \text{fz} \Rightarrow \text{'0'}$ 
  partial  $(\text{wk}\ a)\ (\text{fs}\ i) \Rightarrow \text{wk}(\text{partial}\ a\ i)$ 
partial  $\text{'0'}\ i \Rightarrow \text{'0'}$ 
partial  $(\text{'plus'}\ a\ b)\ i \Rightarrow \text{'plus'}(\text{partial}\ a\ i)\ (\text{partial}\ b\ i)$ 
partial  $\text{'1'}\ i \Rightarrow \text{'0'}$ 
partial  $(\text{'times'}\ a\ b)\ i \Rightarrow$ 
   $\text{'plus'}(\text{'times'}(\text{partial}\ a\ i)\ b)\ (\text{'times'}\ a\ (\text{partial}\ b\ i))$ 
partial  $(\text{def}\ f\ a)\ i \Rightarrow$ 
   $\text{'plus'}(\text{def}(\text{partial}\ f\ (\text{fs}\ i))\ a)\ (\text{'times'}(\text{def}(\text{partial}\ f\ \text{fz})\ a)\ (\text{partial}\ a\ i))$ 
partial  $(\text{'mu'}\ f)\ i \Rightarrow \text{'mu'}(\text{'plus'}(\text{wk}(\text{def}(\text{partial}\ f\ (\text{fs}\ i))\ (\text{'mu'}\ f)))$ 
   $(\text{'times'}(\text{wk}(\text{def}(\text{partial}\ f\ \text{fz})\ (\text{'mu'}\ f)))\ \text{vl})$ 

```

*Exercise 17.* Calculate (by hand) the derivative of rose trees, i.e. the value of **partial fz rt**

## 6.2 Generic plugging

To convince ourselves that the definition of derivatives as one hole contexts given above is correct we derive<sup>10</sup> a generic version of the generic plugging operation:

$$\underline{\text{let}} \frac{a\ i\ x : \text{Elcf}(\text{partial}\ a\ i)\ as\ y : \text{Elcf}(\text{var}\ i)\ as}{\text{gplug}\ a\ i\ x\ y : \text{Elcf}\ a\ as}$$

That is given an element  $x$  of a partial derivative of  $a$  at  $i$  we can fill the hole with an element of the corresponding type of the telescope, obtaining an element of  $a$ .

We construct **gplug** by recursion over  $x$ , however, unlike in the previous examples, which were completely data driven we have to analyse the type directly, i.e. we have to invoke **case**  $a$ . We discuss the cases:

**variable**

$$\begin{aligned} \text{gplug}\ \text{vl}\ \text{fz}\ \text{void}\ y &\Rightarrow y \\ \text{gplug}\ \text{vl}\ (\text{fs}\ i)\ x\ y &\Leftarrow \underline{\text{case}}\ x \end{aligned}$$

If the index is **fz** the argument  $y$  is the filler we are looking for, otherwise the derivative is the empty type and we eliminate it by a vacuous case analysis.

<sup>10</sup> We were unable to convince Epigram to check all of the definition below due to a space leak in the current implementation. We are hopeful that this will be fixed in the next release of Epigram.

## weakening

$$\begin{aligned} \mathbf{gplug} (\mathbf{wk} \ a) \ \mathbf{fz} \ x \ y &\leftarrow \underline{\mathbf{case} \ x} \\ \mathbf{gplug} (\mathbf{wk} \ a) \ (\mathbf{fs} \ i) \ (\mathbf{pop} \ x) \ (\mathbf{pop} \ y) &\Rightarrow \mathbf{pop} \ (\mathbf{gplug} \ a \ i \ x \ y) \end{aligned}$$

This is in some way dual to the previous case: if the index is **fz** we have the empty derivative, otherwise we recur.

## constant types

$$\begin{aligned} \mathbf{gplug} \ '0' \ i \ x \ y &\leftarrow \underline{\mathbf{case} \ x} \\ \mathbf{gplug} \ '1' \ i \ x \ y &\leftarrow \underline{\mathbf{case} \ x} \end{aligned}$$

are easy because impossible, since the derivative is the empty type.

## disjoint union

$$\begin{aligned} \mathbf{gplug} \ ('plus' \ a \ b) \ i \ (\mathbf{inl} \ xa) \ y &\Rightarrow \mathbf{inl} \ (\mathbf{gplug} \ a \ i \ xa \ y) \\ \mathbf{gplug} \ ('plus' \ a \ b) \ i \ (\mathbf{inr} \ xb) \ y &\Rightarrow \mathbf{inr} \ (\mathbf{gplug} \ b \ i \ xb \ y) \end{aligned}$$

the injections are just carried through.

## Product

$$\begin{aligned} \mathbf{gplug} \ ('times' \ a \ b) \ i \ (\mathbf{inl} \ (\mathbf{pair} \ xa \ xb)) &\Rightarrow \mathbf{pair} \ (\mathbf{gplug} \ a \ i \ xa \ y) \ xb \\ \mathbf{gplug} \ ('times' \ a \ b) \ i \ (\mathbf{inr} \ (\mathbf{pair} \ xa \ xb)) &\Rightarrow \mathbf{pair} \ xa \ (\mathbf{gplug} \ b \ i \ xb \ y) \end{aligned}$$

The derivative records the information in which component of the pair we can find the hole.

## Local definition

$$\begin{aligned} \mathbf{gplug} \ (\mathbf{def} \ f \ a) \ i \ (\mathbf{inl} \ (\mathbf{push} \ x)) \ y &\Rightarrow \mathbf{push} \ (\mathbf{gplug} \ f \ (\mathbf{fs} \ i) \ x \ (\mathbf{pop} \ y)) \\ \mathbf{gplug} \ (\mathbf{def} \ f \ a) \ i \ (\mathbf{inr} \ (\mathbf{pair} \ (\mathbf{push} \ x) \ q)) \ y &\Rightarrow \\ &\mathbf{push} \ (\mathbf{gplug} \ f \ \mathbf{fz} \ x \ (\mathbf{top} \ (\mathbf{gplug} \ a \ i \ q \ y))) \end{aligned}$$

In the first case the hole is in top-level ( $f$ ) tree but not at the first variable, which is used in the definition. In the 2nd case the hole is in a subtree ( $a$ ) which means we have to plug the hole there and then use the result to plug a hole in the top-level tree.

*Exercise 18.* Complete (using pen and paper) the definition of **gplug** by implementing the case for **mu**.

## 6.3 Derivatives of containers

Previously, we have defined derivatives by induction over the syntax of types. Using containers we can give a more direct, semantic definition. The basic idea can be related to derivatives of polynomials, i.e. the derivative of  $f \ x = x^n$  is

$f' x = n \times x^{n-1}$ . As a first step we need to find a type-theoretic counterpart to the predecessor of a type by removing one element of the type. We define:

$$\text{data } \frac{A : \star \quad a : A}{\text{Minus } A a : \star} \quad \text{where } \frac{a' : A \quad na : \mathbf{Not} (a = a')}{\text{minus } a na : \text{Minus } A a'}$$

We can embed  $\text{Minus } A a$  back into  $A$ :

$$\text{let } \frac{m : \text{Minus } A a}{\text{emb } m : A} \quad \text{emb } m \leftarrow \text{case } m \\ \text{emb } (\text{minus } a' na) \Rightarrow a'$$

We can analyse  $A$  in terms of  $\text{Minus } A a$  by defining a view. An element of  $A$  is either  $a$  or it is in the range of  $\text{emb}$ :

$$\text{data } \frac{a; a' : A}{\text{MinusV } a a'} \\ \text{where } \frac{}{\text{same } a : \text{MinusV } a a'} \quad \frac{m : \text{Minus } A a}{\text{other } m : \text{MinusV } a (\text{emb } m)}$$

This view is exhaustive, if  $A$  has a decidable equality:

$$\text{let } \frac{a, a' : A \quad eq : \mathbf{Dec} (a = a')}{\text{minusV}' a a' eq : \text{MinusV } a a'} \\ \text{minusV}' a a' eq \leftarrow \text{case } eq \\ \text{minusV}' a a (\text{yes refl}) \Rightarrow \text{same } a \\ \text{minusV}' a a' (\text{no } f) \Rightarrow \text{other } (\text{minus } a' f) \\ \text{let } \frac{eqA : \mathbf{DecEq } A \quad a, a' : A}{\text{minusV } eqA a a' : \text{MinusV } a a'} \\ \text{minusV } eqA a a' \Rightarrow \text{minusV}' a a' (eqA a a')$$

We are now ready to construct the derivative of containers and implement a variant **plug** for containers. To simplify the presentation we first restrict our attention to unary containers.

Given a unary container  $\text{ucont } S P$  its derivative is given by shapes which are the original shapes together with a chosen position, i.e.  $\text{Sigma } S P$ . The new type of positions is obtained by subtracting this chosen element from  $P$  hence we define:

$$\text{let } \frac{P : S \rightarrow \star \quad sp : \text{Sigma } S P}{\text{derivP } P sp : \star} \\ \text{derivP } P sp \leftarrow \text{case } sp \\ \text{derivP } P (\text{tup } s p) \Rightarrow \text{Minus } (P s) p$$

and hence the derivative of a unary container is given by:

$$\text{let } \frac{C : \mathbf{UCont}}{\text{derivC } C : \mathbf{UCont}} \\ \text{derivC } C \leftarrow \text{case } C \\ \text{derivC } (\text{ucont } S P) \Rightarrow \text{ucont } (\text{Sigma } S P) (\text{derivP } P)$$

While the definition above works for any unary container, we need decidability of equality on positions to define the generic plugging operation. Intuitively, we have to be able to differentiate between position to identify the location of a hole. Hence, only containers with a decidable equality are differentiable. We define a predicate on containers:

$$\text{data } \frac{C : \mathbf{UCont}}{\mathbf{DecUCont } C : \star} \text{ where } \frac{decP : \forall s : S \Rightarrow \mathbf{DecEq } (P s)}{\mathbf{decUCont } decP : \mathbf{DecUCont } (\mathbf{ucont } S P)}$$

Given a decidable unary container we can define the function **uplug** which given an element of the extension of the derivative of a container  $x : \mathbf{UExt}(\mathbf{derivC } C) X$  and an element  $y : X$  we can plug the hole with  $y$  thus obtaining an element of  $\mathbf{UExt } C X$ :

$$\begin{aligned} & eq : \mathbf{DecEq } A \quad a : A \\ & f : \mathbf{Minus } A a \rightarrow X \\ & x : X \quad a' : A \\ \text{let } & \frac{}{\mathbf{mplug } eq a f x a' : X} \\ & \mathbf{mplug } eq a f x a' \Leftarrow \text{view } \mathbf{minusV } eq a a' \\ & \mathbf{mplug } eq a f x a \Rightarrow x \\ & \mathbf{mplug } eq a f x (\mathbf{emb } m) \Rightarrow f m \\ \\ & C : \mathbf{UCont} \quad d : \mathbf{DecUCont } C \\ & x : \mathbf{UExt } (\mathbf{derivC } C) X \quad y : X \\ \text{let } & \frac{}{\mathbf{uplug } C d x y : \mathbf{UExt } C X} \\ & \mathbf{uplug } C d x y \Leftarrow \text{case } C \\ & \mathbf{uplug } (\mathbf{ucont } S P) d x y \Leftarrow \text{case } d \\ & \mathbf{uplug } (\mathbf{ucont } S P) (\mathbf{decUCont } decP) x y \Leftarrow \text{case } x \\ & \mathbf{uplug } (\mathbf{ucont } S P) (\mathbf{decUCont } decP) (\mathbf{uext } sp f) y \Leftarrow \text{case } sp \\ & \mathbf{uplug } (\mathbf{ucont } S P) (\mathbf{decUCont } decP) (\mathbf{uext } (\mathbf{tup } s p) f) y \\ & \Rightarrow \mathbf{uext } a (\mathbf{mplug } (decP a) b f y) \end{aligned}$$

*Exercise 19.* Extend the derivative operator for containers to  $n$ -ary containers, i.e. define

$$\text{let } \frac{C : \mathbf{Cont } n \quad i : \mathbf{Fin } n}{\mathbf{partialC } C i : \mathbf{Cont } n}$$

To extend the plug operator we have to define decidability for an  $n$ -ary container. We also need to exploit that equality for finite types is decidable.

## 7 Conclusions and further work

Using dependent types we were able to define different universes and generic operations on them. We have studied two fundamentally different approaches: a semantic approach, first using finite types and the container types and a syntactic approach where the elements are defined inductively. Further work needs

to be done to relate the two more precisely, they are only two views of the same collection of types. We have already observed that there is a trade-off between the size of the universe, i.e. the collection of types definable within it, and the number of generic operations. The previous section suggests that between the context-free types and the strictly positive types: the differentiable types, i.e. the types with a decidable equality on positions. Previous work in a more categorical framework [7] shows already that the types which are obtained by closing context-free types under a coinductive type former ( $\nu$ ) are still differentiable.

The size of a universe is not the only parameter we can vary, the universes we have considered here are still very coarse. E.g. while we have a more refined type system on the meta-level, dependent types, this is not reflected in our universes. We have no names for the family of finite types, the vectors or the family of elements of a universe itself. Recent work shows that it is possible to extend both the syntactic and the semantic approach to capture families of types, see [34, 8]. Another direction to pursue is to allow types where the positions are result of a quotient, like bags or multisets. We have already investigated this direction from a categorical point of view [6]; a typetheoretic approach requires a Type Theory which allows quotient types. Here our current work on *Observational Type Theory* [10] fits in very well.

Apart from the more theoretical questions regarding universes of datatypes there are more pragmatic issues. We don't want to work with isomorphic copies of our datatypes, but we want to be able to access the top-level types themselves. We are working on a new implementation of Epigram which will provide a quotation mechanism which makes the top-level universe accessible for the programmer. We also hope to be able to find a good pragmatic answer to vary the level of genericity, i.e. to be able to define generic operations for the appropriate universe without repeating definitions.

## References

1. Michael Abbott. *Categories of Containers*. PhD thesis, University of Leicester, 2003.
2. Michael Abbott, Thorsten Altenkirch, and Neil Ghani. Categories of containers. In *Proceedings of Foundations of Software Science and Computation Structures*, 2003.
3. Michael Abbott, Thorsten Altenkirch, and Neil Ghani. Representing nested inductive types using W-types. In *Automata, Languages and Programming, 31st International Colloquium (ICALP)*, pages 59 – 71, 2004.
4. Michael Abbott, Thorsten Altenkirch, and Neil Ghani. Containers - constructing strictly positive types. *Theoretical Computer Science*, 342:3–27, September 2005. Applied Semantics: Selected Topics.
5. Michael Abbott, Thorsten Altenkirch, Neil Ghani, and Conor McBride. Derivatives of containers. In *Typed Lambda Calculi and Applications, TLCA*, 2003.
6. Michael Abbott, Thorsten Altenkirch, Neil Ghani, and Conor McBride. Constructing polymorphic programs with quotient types. In *7th International Conference on Mathematics of Program Construction (MPC 2004)*, 2004.

7. Michael Abbott, Thorsten Altenkirch, Neil Ghani, and Conor McBride.  $\partial$  for data. *Fundamentae Informatica*, 65(1,2):1 – 28, March 2005. Special Issue on Typed Lambda Calculi and Applications 2003.
8. Thorsten Altenkirch, Neil Ghani, Peter Hancock, Conor McBride, and Peter Morris. Indexed containers. Manuscript, available online, February 2006.
9. Thorsten Altenkirch and Conor McBride. Generic programming within dependently typed programming. In *Generic Programming*, 2003. Proceedings of the IFIP TC2 Working Conference on Generic Programming, Schloss Dagstuhl, July 2002.
10. Thorsten Altenkirch and Conor McBride. Towards observational type theory. Manuscript, available online, February 2006.
11. Thorsten Altenkirch, Conor McBride, and James McKinna. Why dependent types matter. Manuscript, available online, April 2005.
12. Thorsten Altenkirch, Conor McBride, and Peter Morris. Code for generic programming with dependent types, 2007. <http://www.e-pig.org/downloads/GPwDT>.
13. Thorsten Altenkirch and Tarmo Uustalu. Normalization by evaluation for  $\lambda^{-2}$ . In *Functional and Logic Programming*, number 2998 in LNCS, pages 260 – 275, 2004.
14. Andres Loeh, Johan Jeuring (editors); Dave Clarke, Ralf Hinze, Alexey Rodriguez, Jan de Wit. Generic Haskell User’s Guide – Version 1.42 (Coral). Technical Report UU-CS-2005-004, Institute of Information and Computing Sciences, Utrecht University, 2005.
15. Roland Backhouse, Patrik Jansson, Johan Jeuring, and Lambert Meertens. Generic Programming—An Introduction. In S. Doaitse Sweierstra, Pedro R. Henriques, and José N. Oliveira, editors, *Advanced Functional Programming, Third International Summer School (AFP ’98); Braga, Portugal*, volume 1608 of LNCS, pages 28–115. Springer-Verlag, 1998.
16. M. Benke, P. Dybjer, and P. Jansson. Universes for generic programs and proofs in dependent type theory. *Nordic Journal of Computing*, 10(4):265–289, 2003.
17. Karl Crary, Stephanie Weirich, , and Greg Morrisett. Intensional polymorphism in type erasure semantics. *Journal of Functional Programming*, 12(6):567–600, November 2002.
18. P. Dybjer and A. Setzer. A finite axiomatization of inductive-recursive definitions. *Typed Lambda Calculi and Applications*, 1581:129–146, 1999.
19. P. Dybjer and A. Setzer. Indexed induction–recursion. *Journal of Logic and Algebraic Programming*, 66(1):1–49, 2006.
20. Peter Dybjer. Inductive Sets and Families in Martin-Löf’s Type Theory. In Gérard Huet and Gordon Plotkin, editors, *Logical Frameworks*. CUP, 1991.
21. Ralf Hinze. Generic programs and proofs. Habilitationsschrift, Universitt Bonn, 2000.
22. Ralf Hinze and Andres Löh. “Scrap Your Boilerplate” Revolutions. In Tarmo Uustalu, editor, *Mathematics of Program Construction, 2006*, volume 4014 of LNCS, pages 180–208. Springer-Verlag, 2006.
23. Ralf Hinze, Andres Löh, and Bruno C. D. S. Oliveira. “Scrap Your Boilerplate” Reloaded. In Masami Hagiya and Philip Wadler, editors, *FLOPS*, volume 3945 of *Lecture Notes in Computer Science*, pages 13–29. Springer, 2006.
24. Gérard Huet. The Zipper. *Journal of Functional Programming*, 7(5):549–554, 1997.
25. Andres Löh. *Exploring Generic Haskell*. PhD thesis, Utrecht University, Netherlands, September 2004.



26. Zhaohui Luo and Robert Pollack. LEGO Proof Development System: User's Manual. Technical Report ECS-LFCS-92-211, Laboratory for Foundations of Computer Science, University of Edinburgh, 1992.
27. Lena Magnusson and Bengt Nordström. The ALF proof editor and its proof engine. In Henk Barendregt and Tobias Nipkow, editors, *Types for Proofs and Programs*, LNCS 806. Springer-Verlag, 1994. Selected papers from the Int. Workshop TYPES '93, Nijmegen, May 1993.
28. Per Martin-Löf. *Intuitionistic Type Theory*. Bibliopolis-Napoli, 1984.
29. Conor McBride. The Derivative of a Regular Type is its Type of One-Hole Contexts. Available online, 2001.
30. Conor McBride. Epigram, 2004. <http://www.e-pig.org/>.
31. Conor McBride. Epigram: Practical programming with dependent types. In Varmo Vene and Tarmo Uustalu, editors, *Advanced Functional Programming 2004*, Lecture Notes in Computer Science. Springer-Verlag, 2005+. Revised lecture notes from the International Summer School in Tartu, Estonia.
32. Conor McBride and James McKinna. The view from the left. *Journal of Functional Programming*, 14(1), 2004.
33. Fred McBride. *Computer Aided Manipulation of Symbols*. PhD thesis, Queen's University of Belfast, 1970.
34. Peter Morris, Thorsten Altenkirch, and Neil Ghani. Constructing strictly positive families. In *The Australasian Theory Symposium (CATS2007)*, January 2007.
35. Peter Morris, Thorsten Altenkirch, and Conor McBride. Exploring the regular tree types. In Christine Paulin-Mohring Jean-Christophe Filliatre and Benjamin Werner, editors, *Types for Proofs and Programs (TYPES 2004)*, Lecture Notes in Computer Science, 2006.
36. H. Pfeifer and H. Rueß. Polytypic abstraction in type theory. In Roland Backhouse and Tim Sheard, editors, *Workshop on Generic Programming (WGP'98)*. Dept. of Computing Science, Chalmers Univ. of Techn. and Göteborg Univ., June 1998.
37. Bengt Nordström, Kent Petersson, and Jan Smith. *Programming in Martin-Löf's type theory: an introduction*. Oxford University Press, 1990.
38. Dimitrios Vytiniotis, Stephanie Weirich, and Simon Peyton Jones. Boxy type inference for higher-rank types and impredicativity. In *Proceedings of the International Conference on Functional Programming (ICFP) 2006*, September 2006.
39. Philip Wadler. Views: A way for pattern matching to cohabit with data abstraction. In *Proceedings of POPL '87*. ACM, 1987.
40. Stephanie Weirich. RepLib: A library for derivable type classes. In Andres Löf, editor, *Proceedings of the ACM Haskell Workshop, 2006*, 2006.